code:
;-------------------------------------------------
;상대주도정상위
;섹스계 커맨드, 레벨 1
;-------------------------------------------------
@COM320
;-------------------------------------------------
;특수 파생 처리
;-------------------------------------------------

;침대가 없으면 립 정도에
SIF !BEDROOM(CFLAG:PLAYER:현재위치) && !(BATHROOM() && ITEM:20)
	TFLAG:193 = 1

;-------------------------------------------------
;COM별 FLAG의 설정
;-------------------------------------------------
;행위자
TCVAR:116 = PLAYER

;봉사 쾌락 플래그
TCVAR:PLAYER:108 = 1

CALL TOUCH_SET(SET_FROM_V, 1, TARGET)
TCVAR:PLAYER:2 = 1
TCVAR:PLAYER:동정상실 = 1
TEQUIP:TARGET:체위 = 1
TFLAG:MASTER의P삽입중 = 1
TEQUIP:50 = PLAYER

CSTR:PLAYER:10 += @"{TARGET}/"
;TCVAR:2 사정장소 플래그(-1=콘돔 1=질내 2=애널 3=수음 4=펠라티오 5=파이즈리 6=가랑이 7=풋잡 8=체표 9=애널 봉사 10=도구 11=촉수 12=A고문 13=V 꾸짖고)
TCVAR:PLAYER:2 = 1
TCVAR:PLAYER:동정상실 = 1
;강도
TCVAR:PLAYER:4 = 800

CSTR:TARGET:10 += @"{PLAYER}/"
;TCVAR:2 사정장소 플래그(-1=콘돔 1=질내 2=애널 3=수음 4=펠라티오 5=파이즈리 6=가랑이 7=풋잡 8=체표 9=애널 봉사 10=도구 11=촉수 12=A고문 13=V 꾸짖고)
TCVAR:TARGET:2 = 13
;강도
TCVAR:TARGET:4 = 1500

;정상위 보정
;PLAYER가 TARGET의 질에 삽입하고 있다
SIF (BEDROOM(CFLAG:PLAYER:현재위치) || (BATHROOM() && ITEM:20)) && TEQUIP:50 == PLAYER
	TCVAR:PLAYER:4 += EXP:애정경험 * 5
SIF TCVAR:PLAYER:4 >= 3000
	TCVAR:PLAYER:4 = 3000

DOWNBASE:PLAYER:체력 += 30
DOWNBASE:PLAYER:기력 += 40

SOURCE:PLAYER:쾌C = 400

DOWNBASE:체력 += 40
DOWNBASE:기력 += 50

SOURCE:PLAYER:성행동+= 20
SOURCE:쾌V = 400
SOURCE:사랑 = 300
SOURCE:고통 = 500
SOURCE:노출 = 50
SIF TFLAG:193
	SOURCE:노출 = 300
;정상위 보정
IF BEDROOM(CFLAG:PLAYER:현재위치) && !ASSIPLAY
	SOURCE:PLAYER:성행동+= EXP:애정경험/20
	SOURCE:쾌V += EXP:애정경험
ENDIF
IF SOURCE:쾌V >= 1000
	SOURCE:PLAYER:성행동+= 50
	SOURCE:쾌V = 1000
ENDIF

;위 쪽으로 계산한 더러운 데이터
SOURCE:불결 = Y*10 + 60
CALL 특수소스처리
;V에의 고통은 먼저 처리해 둔다
CALL PAIN_CHECK_V(TARGET)
;-------------------------------------------------
;더러움의 처리
;-------------------------------------------------
;사정에 관계없이 행해지는 처리는 여기에서
CALL STAIN_질내P삽입
;-------------------------------------------------
;경험의 처리
;-------------------------------------------------
EXP:PLAYER:삽입경험 ++
EXP:V경험 ++
;동성애 레즈비언이라면 반환값 1 동성애자라면 반환값 2

IF HETEROSEX(TARGET, PLAYER)
	EXP:(50 + HETEROSEX(TARGET, PLAYER)) += 4
	EXP:PLAYER:(50 + HETEROSEX(TARGET, PLAYER)) += 4
ENDIF
EXP:V성교경험 ++
IF !ASSIPLAY
	EXP:애정경험 += TALENT:연모 * 3
	;처녀
	SIF TALENT:처녀 == 1 && TALENT:연모
		EXP:애정경험 += 10
ENDIF

RETURN 1

;-------------------------------------------------
;실행 판정
;-------------------------------------------------
@COM_ABLE320
SIF !TFLAG:100
	RETURN 0
SIF TCVAR:20 != 41
	RETURN 0
;조교자가 남자, 양성의 경우는 어느 정도 발기하고 있지 않으면 안 돼
SIF (TALENT:PLAYER:2 & 2) && BASE:PLAYER:발기 < 1000
	RETURN 0
;바이브 삽입중은 안 돼
SIF TEQUIP:13
	RETURN 0
;남자라면 안 돼
SIF !(TALENT:2 & 1)
	RETURN 0
;체격 차이가 있는 경우는 경험이 없으면 안 돼
SIF 삽입불가(TARGET) == 1 && EXP:V경험 < 50 && !TALENT:PLAYER:금단의지식
	RETURN 0
;한층 더 체격 차이가 있으면(자) 금단의지식이 필요
SIF 삽입불가(TARGET) == 2 && (!TALENT:PLAYER:금단의지식 || EXP:V경험 < 50)
	RETURN 0
SIF 삽입불가(TARGET) == 3
	RETURN 0
;조교자가 남자, 반인반요, 후타나리, 혹은 페니스밴드가 없으면 안 돼
SIF !(TALENT:PLAYER:2 & 2) && ITEM:PBAND == 0
	RETURN 0
;처녀로, 순종 4 이하·레즈끼 4 이하의 ASSI의 경우는 할 수 없다(새드라면 OK)
IF TALENT:처녀 && ASSIPLAY
	SIF (ABL:ASSI:순종 <= 4 || ABL:ASSI:레즈끼 <= 4) && TALENT:ASSI:새드 == 0
		RETURN 0
ENDIF
;시간정지중은 스킵
IF FLAG:70 != 1
	;처녀 또한 윤활 부족은 안 돼
	SIF PALAM:윤활 < PALAMLV:2 && TALENT:처녀 && TALENT:처녀 != -1
		RETURN 0
ENDIF

;하반신착의
SIF TEQUIP:0 & 4
	RETURN 0
;목욕탕플레이중은 플레이매트가 없으면 안 돼
IF TEQUIP:31
	SIF ITEM:20 == 0 && NOITEM == 0
		RETURN 0
ENDIF
;후배위로부터는 안 돼
SIF TEQUIP:체위 == 5
	RETURN 0
RETURN 1

path: ERB/커맨드관련/COMF/COMF320 상대주도정상위_t.ERB


repo_name: purugi/eraTWKR

path: ERB/커맨드관련/COMF/COMF322 상대주도후배위_t.ERB

license: mit


path: ERB/커맨드관련/COMF/COMF323 상대주도후배위Ａ_t.ERB


path: ERB/커맨드관련/COMF/COMF4 애널핥기_t.ERB


path: ERB/커맨드관련/COMF/COMF40 로터_t.ERB


path: ERB/커맨드관련/COMF/COMF41 E마사지기_t.ERB


path: ERB/커맨드관련/COMF/COMF42 클리캡_t.ERB


path: ERB/커맨드관련/COMF/COMF43 오나홀_t.ERB


code:
;-------------------------------------------------
;바이브
;도구 사용 커맨드, 레벨 2
;-------------------------------------------------
@COM44
;착탈의처리
TEQUIP:바이브 = !TEQUIP:바이브
;-------------------------------------------------
;COM별 FLAG의 설정
;-------------------------------------------------
;행위자
TCVAR:116 = PLAYER
;봉사 쾌락 플래그
TCVAR:PLAYER:108 = 1

CSTR:TARGET:10 += @"{PLAYER}/"
;TCVAR:2 사정장소 플래그(-1=콘돔 1=질내 2=애널 3=수음 4=펠라티오 5=파이즈리 6=가랑이 7=풋잡 8=체표 9=애널 봉사 10=도구 11=촉수 12=A고문 13=V 꾸짖고)
TCVAR:TARGET:2 = 13
;강도
TCVAR:TARGET:4 = 900

;-------------------------------------------------
;소스 추가 처리
;-------------------------------------------------
;소비 체력, 기력
IF FLAG:70 == 1
	BASE:MASTER:TSP -= 100
SOURCE:PLAYER:성행동+= 10
	SOURCE:쾌V = 250
	SOURCE:고통 = 25

	;처녀였던 경우는, 고통의 소스와 반발의 소스를 추가한다
	IF TALENT:처녀 == 1
		SOURCE:고통 += 1000
	ENDIF

	;재생 처녀였던 경우는, 고통의 소스와 반발의 소스를 추가한다
	IF TALENT:처녀 == 2
		SOURCE:고통 += 250
	ENDIF

ELSE

	DOWNBASE:체력 += 30
	DOWNBASE:기력 += 100
SOURCE:PLAYER:성행동+= 10

	SOURCE:쾌V = 250
	SOURCE:고통 = 50
	SOURCE:노출 = 20
	SOURCE:일탈 = 40
	SOURCE:반감 = 40

	;-------------------------------------------------
	;특수소스처리
	;-------------------------------------------------
	;억압, 저항 소유에는, 억울소스 추가
	SIF TALENT:자기애 < 0 || TALENT:저항
		SOURCE:울굴 += 40

	;처녀였던 경우는, 고통의 소스와 반발의 소스를 추가한다
	IF TALENT:처녀 == 1
		SOURCE:고통 += 2000
		SOURCE:반감 += 3000
	ENDIF

	;재생 처녀였던 경우는, 고통의 소스와 반발의 소스를 추가한다
	IF TALENT:처녀 == 2
		SOURCE:고통 += 500
		SOURCE:반감 += 1000
	ENDIF
ENDIF
;V에의 고통은 먼저 처리해 둔다
CALL PAIN_CHECK_V(TARGET)

;-------------------------------------------------
;경험의 처리
;-------------------------------------------------
;삽입시는 V경험+
EXP:V경험 ++
IF FLAG:70 == 1
	IF HETEROSEX(TARGET, PLAYER)
		EXP:PLAYER:(50 + HETEROSEX(TARGET, PLAYER)) += 1
	ENDIF
ELSE
	;이상경험
	IF TALENT:처녀 == 1 && !GETBIT(CFLAG:3, 3)
		EXP:이상경험 ++
		SETBIT CFLAG:3, 3
	ENDIF

	;동성애 레즈비언이라면 반환값 1 동성애자라면 반환값 2
	IF HETEROSEX(TARGET, PLAYER)
		EXP:(50 + HETEROSEX(TARGET, PLAYER)) += 1
		EXP:PLAYER:(50 + HETEROSEX(TARGET, PLAYER)) += 1
	ENDIF
ENDIF
EXP:PLAYER:도구사용경험 ++

RETURN 1

;-------------------------------------------------
;장착 계속의 처리
;-------------------------------------------------
@EQUIP_COM44(ARG)
;바이브 삽입중
PRINTL ＜바이브 삽입중＞

CSTR:TARGET:10 += @"{PLAYER}/"
TCVAR:TARGET:2 = 13
TCVAR:TARGET:4 = 450

IF FLAG:70 == 1
BASE:MASTER:TSP -= 50
SOURCE:ARG:쾌V += 125

ELSE

DOWNBASE:ARG:체력 += 10
DOWNBASE:ARG:기력 += 50

SOURCE:ARG:쾌V += 125
SOURCE:ARG:노출 += 10
SOURCE:ARG:일탈 += 20

;-------------------------------------------------
;특수소스처리
;-------------------------------------------------
;억압, 저항 소유에는, 억울소스 추가
SIF TALENT:ARG:자기애 < 0 || TALENT:ARG:저항
	SOURCE:ARG:울굴 += 20

;V에의 고통은 먼저 처리해 둔다
;EXP:V경험을 본다(처녀는 있을 수 없다)
LOCAL = 25
LOCAL:1 = 20
IF EXP:ARG:V경험 < EXPLV:2
	TIMES LOCAL , 0.80
	TIMES LOCAL:1 , 1.20
ELSEIF EXP:ARG:V경험 < EXPLV:3
	TIMES LOCAL , 0.40
ELSEIF EXP:ARG:V경험 < EXPLV:4
	TIMES LOCAL , 0.10
ELSEIF EXP:ARG:V경험 < EXPLV:5
	TIMES LOCAL , 0.00
ELSEIF EXP:ARG:V경험 >= EXPLV:5
	TIMES LOCAL , 0.00
ENDIF

;PALAM:윤활을 본다
IF PALAM:ARG:윤활 < PALAMLV:1
	TIMES LOCAL , 1.20
	TIMES LOCAL:1 , 1.50
ELSEIF PALAM:ARG:윤활 < PALAMLV:2
	TIMES LOCAL , 0.70
	TIMES LOCAL:1 , 1.20
ELSEIF PALAM:ARG:윤활 < PALAMLV:3
	TIMES LOCAL , 0.40
ELSEIF PALAM:ARG:윤활 < PALAMLV:4
	TIMES LOCAL , 0.10
ELSEIF PALAM:ARG:윤활 >= PALAMLV:4
	TIMES LOCAL , 0.10
ENDIF
SOURCE:ARG:고통 += LOCAL
SOURCE:ARG:반감 += LOCAL:1

;-------------------------------------------------
;경험의 처리
;-------------------------------------------------
;자위 실행시는 경험＋1
SIF EXP_UP(22, TARGET)
	EXP:ARG:V경험 += 1
ENDIF
RETURN 1

;-------------------------------------------------
;실행 판정
;-------------------------------------------------
@COM_ABLE44
;바이브 실행 판정
SIF !TFLAG:100
	RETURN 0
;일괄관리
SIF GLOBAL_COMABLE(44)
	RETURN RESULT
;해제는 언제라도 가능
SIF TEQUIP:13
	RETURN 1
;남자라면 안 돼
SIF !(TALENT:2 & 1)
	RETURN 0
;아이템 을 가지고 있는인가, 아이템 없음 설정이 아니면 안 돼
SIF ITEM:4 == 0 && NOITEM == 0
	RETURN 0

;하반신착의
SIF TEQUIP:0 & 4
	RETURN 0
;처녀로, 순종 4 이하·레즈끼 4 이하의 ASSI의 경우 바이브는 할 수 없다(새드라면 OK)
IF TALENT:처녀 && ASSIPLAY
	SIF (ABL:ASSI:순종 <= 4 || ABL:ASSI:레즈끼 <= 4) && TALENT:ASSI:새드 == 0
		RETURN 0
ENDIF
;윤활 부족해, 순종 3 이하·레즈끼 3 이하의 ASSI의 경우 바이브는 할 수 없다(새드라면 OK)
IF PALAM:윤활 < PALAMLV:2 && ASSIPLAY
	SIF (ABL:ASSI:순종 <= 3 || ABL:ASSI:레즈끼 <= 3) && TALENT:ASSI:새드 == 0
		RETURN 0
ENDIF
RETURN 1

path: ERB/커맨드관련/COMF/COMF44 바이브_t.ERB


path: ERB/커맨드관련/COMF/COMF45 애널바이브_t.ERB


path: ERB/커맨드관련/COMF/COMF46 애널비즈_t.ERB


path: ERB/커맨드관련/COMF/COMF47 유두캡_t.ERB


path: ERB/커맨드관련/COMF/COMF48 착유기_t.ERB


path: ERB/커맨드관련/COMF/COMF49 유두로터_t.ERB


code:
;----------------------
;491:オナニーする
;----------------------
;練習に作ってみた簡易オナニー
@COM491
;P持ちでかつ精力か勃起が0だとシコれない。無勃起でイケる人はイケるけど当座は発散目的ということで
IF TALENT:PLAYER:2 != 1 && BASE:MASTER:발기 <=1
	PRINTFORMW 무스코가 풀이 죽어 힘이 나지 않는다
	RETURN -1

ELSEIF TALENT:PLAYER:2 != 1 &&  BASE:MASTER:정력 <=1
	PRINTFORMW 총알 조각으로 의지가 나오지 않는다
	RETURN -1

ELSE
	PRINTFORML %CALLNAME:MASTER%(은)는 적당한 그늘을 찾아내 깨끗이 하기로 했다

ENDIF

;-------------------------------------------------
;COM別FLAGの設定(日常系317,腰を動かすから流用）
;-------------------------------------------------
;行為者
TCVAR:116 = PLAYER

CSTR:PLAYER:10 += @"{PLAYER}/"
;TCVAR:2 射精箇所フラグ（-1=コンドーム 1=膣内 2=アナル 3=手淫 4=口淫 5=パイズリ 6=素股 7=足コキ 8=体表 9=アナル奉仕 10=道具 11=触手 12=Ａ責め 13=Ｖ責め）
TCVAR:PLAYER:2 = 3
;強度
TCVAR:PLAYER:4 = 5000 + ABL:손가락 * 100

;-------------------------------------------------
;ソース追加処理。
;-------------------------------------------------
;消費体力、気力
IF FLAG:70 == 1
	BASE:MASTER:TSP -= 30

ELSE

	DOWNBASE:0 += 200
	DOWNBASE:1 += 300

;精力が減少
;	SIF TALENT:PLAYER:2 != 1
;		BASE:MASTER:정력 -= 50

ENDIF
;勃起が減少
;SIF TALENT:PLAYER:2 != 1
;BASE:MASTER:발기 -=100

;先に時間経過
SIF FLAG:70 == 0
	TIME += 20

;快楽(313,胸愛撫から流用）

	SOURCE:0 = 1000
	IF ABL:PLAYER:손가락 >= 4
		SOURCE:0 += ABL:PLAYER:손가락 * 100
	ELSE
		SOURCE:0 += ABL:PLAYER:손가락 * 500
	ENDIF
	SOURCE:20 = 100
	SOURCE:15 = 50
	SOURCE:17 = 70

;各種経験
EXP:PLAYER:0 += 1
;EXP:PLAYER:10 += 1
;EXP:PLAYER:5 += 1
EXP:PLAYER:22 += 1
EXP:PLAYER:24 += 1

;P持ちなら射精経験
;SIF TALENT:PLAYER:2 != 1
;	EXP:PLAYER:11 += 1

RETURN 1

;-------------------------------------------------
;汚れの処理
;-------------------------------------------------
;あなたの指にあなたのPかVの汚れが移動
;ふたなりなら両方

IF TALENT:PLAYER:2 ==1
	STAIN:PLAYER:1 |= STAIN:PLAYER:3

ELSEIF TALENT:PLAYER:2 ==2
	STAIN:PLAYER:1 |= STAIN:PLAYER:2

ELSE
	STAIN:PLAYER:1 |= STAIN:PLAYER:2
	STAIN:PLAYER:1 |= STAIN:PLAYER:3

ENDIF

;-------------------------------------------------
;実行判定。80行目までと84行目はERB700からコピペ
;-------------------------------------------------
@COM_ABLE491
;目撃者がいると駄目
SIF TARGET > 0 && !FLAG:70
	RETURN 0
SIF !TFLAG:100
	RETURN 0
;一括管理
SIF GLOBAL_COMABLE(491)
	RETURN RESULT
;気力0
SIF BASE:MASTER:기력 <= 0
	RETURN 0
SIF CFLAG:MASTER:지금바쁨
	RETURN 0

RETURN 1

path: ERB/커맨드관련/COMF/COMF491 자위하기.ERB


path: ERB/커맨드관련/COMF/COMF5 애널애무_t.ERB


path: ERB/커맨드관련/COMF/COMF50 콘돔장착_t.ERB


path: ERB/커맨드관련/COMF/COMF51 피임약_t.ERB


path: ERB/커맨드관련/COMF/COMF511 더블파이즈리_t.ERB


path: ERB/커맨드관련/COMF/COMF6 가슴애무_t.ERB


code:
;-------------------------------------------------
;정상위
;섹스계 커맨드, 레벨 1
;-------------------------------------------------
@COM60
;-------------------------------------------------
;특수 파생 처리
;-------------------------------------------------

;전회의 조교가 3 P 때 3 P에
IF PREVCOM == 71
	CALL COM_ABLE71
	SIF RESULT == 1
		JUMP COM71
ENDIF
;침대가 없는 OR욕실에서 플레이매트가 없으면 립 정도에
SIF !BEDROOM(CFLAG:PLAYER:현재위치) && !(BATHROOM() && ITEM:20)
	TFLAG:193 = 1

;-------------------------------------------------
;COM별 FLAG의 설정
;-------------------------------------------------
;행위자
TCVAR:116 = PLAYER
;봉사 쾌락 플래그
TCVAR:PLAYER:108 = 1

CSTR:PLAYER:10 += @"{TARGET}/"
;TCVAR:2 사정장소 플래그(-1=콘돔 1=질내 2=애널 3=수음 4=펠라티오 5=파이즈리 6=가랑이 7=풋잡 8=체표 9=애널 봉사 10=도구 11=촉수 12=A고문 13=V 꾸짖고)
TCVAR:PLAYER:2 = 1
TCVAR:PLAYER:동정상실 = 1
;강도
TCVAR:PLAYER:4 = 1500

CSTR:TARGET:10 += @"{PLAYER}/"
;TCVAR:2 사정장소 플래그(-1=콘돔 1=질내 2=애널 3=수음 4=펠라티오 5=파이즈리 6=가랑이 7=풋잡 8=체표 9=애널 봉사 10=도구 11=촉수 12=A고문 13=V 꾸짖고)
TCVAR:TARGET:2 = 13
;강도
TCVAR:TARGET:4 = 1500

;신설정안
;강도 우선 2할
;TCVAR:PLAYER:4 = 300
;소스는 반강으로
;	SOURCE:쾌V = 200
;	SOURCE:고통 = 130(250)

DOWNBASE:PLAYER:체력 += 40
DOWNBASE:PLAYER:기력 += 40
;-------------------------------------------------
;소스 추가 처리
;-------------------------------------------------
;소비 체력, 기력
IF FLAG:70 == 1
	BASE:MASTER:TSP -= 100
	SOURCE:PLAYER:쾌C = 400
	SOURCE:PLAYER:성행동+= 20
	SOURCE:쾌V = 400
	SOURCE:고통 = 250

	;처녀였던 경우는, 고통의 소스와 반발의 소스를 추가한다
	IF TALENT:처녀 == 1
		SOURCE:고통 += 1000
	ENDIF

	;재생 처녀였던 경우는, 고통의 소스와 반발의 소스를 추가한다
	IF TALENT:처녀 == 2
		SOURCE:고통 += 250
	ENDIF

ELSE

	DOWNBASE:체력 += 50
	DOWNBASE:기력 += 60
	SOURCE:PLAYER:성행동+= 20

	SOURCE:PLAYER:쾌C = 400
	SOURCE:쾌V = 300
	SOURCE:사랑 = 150
	SOURCE:고통 = 500
	SOURCE:노출 = 50
	SIF TFLAG:193
		SOURCE:노출 = 300
	SOURCE:반감 = 300

	;정상위 보정
	;PLAYER가 TARGET의 질에 삽입하고 있다
	IF BEDROOM(CFLAG:PLAYER:현재위치) && TEQUIP:50 == PLAYER
		TCVAR:PLAYER:4 += EXP:애정경험 * 5
		SOURCE:쾌V += EXP:애정경험
		SIF SOURCE:쾌V >= 700
			SOURCE:쾌V = 700
		SIF TCVAR:PLAYER:4 >= 3000
			TCVAR:PLAYER:4 = 3000
	ENDIF

	;위 쪽으로 계산한 더러운 데이터
	SOURCE:불결 = Y*10 + 60
	CALL 특수소스처리
ENDIF
;V에의 고통은 먼저 처리해 둔다
CALL PAIN_CHECK_V(TARGET)
;-------------------------------------------------
;더러움의 처리
;-------------------------------------------------
;사정에 관계없이 행해지는 처리는 여기에서
CALL STAIN_질내P삽입

;-------------------------------------------------
;경험의 처리
;-------------------------------------------------
EXP:PLAYER:삽입경험 ++
EXP:V경험 ++
;동성애 레즈비언이라면 반환값 1 동성애자라면 반환값 2
IF FLAG:70 == 1 || CFLAG:장난
	IF HETEROSEX(TARGET, PLAYER)
		EXP:PLAYER:(50 + HETEROSEX(TARGET, PLAYER)) += 4
	ENDIF
	IF CFLAG:PLAYER:장난 == 2
		EXP:PLAYER:수면간경험 += 1
		EXP:TARGET:수면간경험 += 1
	ENDIF
	SIF !CFLAG:장난
	EXP:PLAYER:시간경험 += 1
ELSE
	IF HETEROSEX(TARGET, PLAYER)
		EXP:(50 + HETEROSEX(TARGET, PLAYER)) += 4
		EXP:PLAYER:(50 + HETEROSEX(TARGET, PLAYER)) += 4
	ENDIF
	EXP:V성교경험 ++
	IF !ASSIPLAY
		EXP:애정경험 += TALENT:연모 * 2
		;처녀
		SIF TALENT:처녀 == 1 && TALENT:연모
			EXP:애정경험 += 10
	ENDIF
ENDIF
RETURN 1

@CAN_COM60(ARG)
;-------------------------------------------------
;실행 판정 처리
;-------------------------------------------------
CALL AUTO_SKIN_EQUIP
IF FLAG:70 == 1 || CFLAG:장난
	SKIPDISP ARG
	IF CFLAG:PLAYER:현재위치 == 10 || CFLAG:PLAYER:현재위치 == 40
		PRINTL 정상위로 범한다
	ELSEIF !BEDROOM(CFLAG:PLAYER:현재위치) && !(BATHROOM() && ITEM:20)
		PRINTL 대면입위로 범한다
	ELSE
		PRINTL 정상위로 범한다
	ENDIF
ELSE
	;밧줄로 묶여져 있을 경우는 무시(밧줄이 없을 때만 실행 판정을 실시한다)
	IF TEQUIP:밧줄 == 0
		SKIPDISP ARG
		IF CFLAG:PLAYER:현재위치 == 10 || CFLAG:PLAYER:현재위치 == 40
			PRINTL 정상위
		ELSEIF !BEDROOM(CFLAG:PLAYER:현재위치) && !(BATHROOM() && ITEM:20)
			PRINTL 대면입위
		ELSE
			PRINTL 정상위
		ENDIF
		A = 0
		S = 0
		;모든 명령에 공통의 요소를 고려
		;(순종이 높다고 명령에 따르기 쉬운 등)
		CALL COM_ORDER

		;ABL:욕망
		IF ABL:욕망
			SIF S
				PRINT  +
			A += ABL:욕망 * 3
			PRINT 욕망
			PRINTV 'LV, ABL:욕망, '(, ABL:욕망 * 3, ')
			S = 1
		ENDIF
		;ABL:V감각
		IF ABL:V감각
			SIF S
				PRINT  +
			A += ABL:V감각 * 2
			PRINT V감각
			PRINTV 'LV, ABL:V감각, '(, ABL:V감각 * 2, ')
			S = 1
		ENDIF
		;쾌락각인
		IF MARK:쾌락각인
			SIF S
				PRINT  +
			A += MARK:쾌락각인 * 3
			PRINT 쾌락각인
			PRINTV 'LV, MARK:쾌락각인, '(, MARK:쾌락각인 * 3, ')
			S = 1
		ENDIF
		;PALAM:윤활(의 부족)
		IF PALAM:윤활 < PALAMLV:3
			PRINT  -
			A -= 5
			PRINT 윤활 부족
			PRINTV '(, 5, ')
			S = 1
		ENDIF
		;PALAM:욕정
		IF GETPALAMLV(PALAM:욕정, 5)
			SIF S
				PRINT  +
			A += GETPALAMLV(PALAM:욕정, 5) * 3
			PRINT 욕정
			PRINTV 'LV, GETPALAMLV(PALAM:욕정, 5), '(, GETPALAMLV(PALAM:욕정, 5) * 3, ')
			S = 1
		ENDIF
		;수줍음
		IF TALENT:수치심 > 0
			PRINT  -
			A -= 2
			PRINT 수줍음
			PRINTV '(, 2, ')
			S = 1
		ENDIF
		;쾌감에 부정
		IF TALENT:쾌감응답 < 0 && !TALENT:연모
			PRINT  -
			A -= 5
			PRINT 쾌감에 부정
			PRINTV '(, 5, ')
			S = 1
		ENDIF
		;남성기피(연모, 친애로 무효)
		IF OTOKOGIRAI(TARGET)
			PRINT  -
			A -= 12
			PRINT 남성기피
			PRINTV '(, 12, ')
			S = 1
		ENDIF
		;연모
		IF TALENT:연모 && ASSIPLAY == 0
			SIF S
				PRINT  +
			A += 15
			PRINT 연모
			PRINTV '(, 15, ')
			S = 15
		ENDIF
		;친애
		IF TALENT:연모 > 1 && ASSIPLAY == 0
			SIF S
				PRINT  +
			A += 5
			PRINT 친애
			PRINTV '(, 5, ')
			S = 5
		ENDIF
		;처녀
		LOCAL = MAX(10,40 - EXP:V경험)
		IF TALENT:처녀
			PRINT  -
			A -= LOCAL
			PRINT 처녀
			PRINTV '(, LOCAL, ')
			S = 1
		;V경험
		ELSEIF EXP:V경험 < EXPLV:2
			PRINT  -
			A -= 15
			PRINT V경험
			PRINT 부족
			PRINTV '(, 15, ')
			S = 1
		ENDIF

		;조교자가 후타나리
		IF (TALENT:PLAYER:2 & 3) == 3
			SIF S
				PRINT  +
			A += 4
			PRINT 후타나리
			PRINTV '(, 4, ')
			S = 1
		ENDIF

		;미약
		IF TCVAR:춘약
			SIF S
				PRINT  +
			A += 6
			PRINT 춘약
			PRINTV '(, 6, ')
			S = 1
		ENDIF

		Y = 0
		;애액의 더러움
		SIF STAIN:PLAYER:2 & 1
			Y += 1
		;정액의 더러움
		SIF STAIN:PLAYER:2 & 4
			Y += 3
		;애널의 더러움
		SIF STAIN:PLAYER:2 & 8
			Y += 7
		SIF STAIN:PLAYER:2 & 16
			Y += 15
		Y = Y * (2 - TALENT:악취내성) / 2

		;서로 더러워져
		IF Y
			PRINT  -
			A -= Y
			PRINT 서로 더러워져
			;악취 둔감
			IF TALENT:악취내성 > 0
				PRINT , 악취 둔감
			ELSEIF TALENT:악취내성 < 0
				PRINT , 악취 민감
			ENDIF
			PRINTV '(, Y, ')
			S = 1
		ENDIF

		;합계를 표시(20이상으로 실행)
		PRINT  =
		PRINTV A

		V = 20
		SIF A < V
			PRINT  <
		SIF A == V
			PRINT  =
		SIF A > V
			PRINT  >
		PRINT 실행치
		PRINTV V

	ELSE
		SKIPDISP ARG
		IF !BEDROOM(CFLAG:PLAYER:현재위치) && !(BATHROOM() && ITEM:20)
			PRINTL 속박 대면입위
		ELSE
			PRINTL 속박 정상위
		ENDIF
	ENDIF
	WAIT
	SKIPDISP 0
	;실행할 수 없다
	SIF A < V
		RETURN 0

	;그외제원 판정
	IF ARG == 1
		SIF TFLAG:17 != TARGET
			TFLAG:16 = 0
		;첫체험 허락할까 판정
		IF GROUPMATCH(TALENT:TARGET:처녀, 1, -1)
			CALL LOST_VIRGIN_STOP(TARGET)
			SIF RESULT
				RETURN 0
		ENDIF
		;위험일 피임요구 판정
		CALL 피임요구(TARGET)
		IF TFLAG:16 && !(TEQUIP:PLAYER:콘돔 || TCVAR:피임약)
			TFLAG:16 -= 1
			RETURN 0
		ENDIF
	ENDIF
ENDIF
SKIPDISP 0
RETURN 1

;-------------------------------------------------
;실행 판정
;-------------------------------------------------
@COM_ABLE60
;정상위 실행 판정
SIF !TFLAG:100
	RETURN 0
;일괄관리
SIF GLOBAL_COMABLE(60)
	RETURN RESULT
;조교자가 남자, 후타나리의 경우는 어느 정도 발기하고 있지 않으면 안 돼
SIF (TALENT:PLAYER:2 & 2) && BASE:PLAYER:발기 < 1000
	RETURN 0
;바이브 삽입중은 안 돼
SIF TEQUIP:13
	RETURN 0
;남자라면 안 돼
SIF !(TALENT:2 & 1)
	RETURN 0
;체격 차이가 있는 경우는 경험이 없으면 안 돼
SIF 삽입불가(TARGET) == 1 && EXP:V경험 < 50 && !TALENT:PLAYER:금단의지식
	RETURN 0
;한층 더 체격 차이가 있으면(자) 금단의지식이 필요
SIF 삽입불가(TARGET) == 2 && (!TALENT:PLAYER:금단의지식 || EXP:V경험 < 50)
	RETURN 0
SIF 삽입불가(TARGET) == 3
	RETURN 0
;조교자가 남자, 반인반요, 후타나리, 혹은 페니스밴드가 없으면 안 돼
SIF !(TALENT:PLAYER:2 & 2) && ITEM:PBAND == 0
	RETURN 0
;처녀로, 순종 4 이하·레즈끼 4 이하의 ASSI의 경우는 할 수 없다(새드라면 OK)
IF TALENT:처녀 && ASSIPLAY
	SIF (ABL:ASSI:순종 <= 4 || ABL:ASSI:레즈끼 <= 4) && TALENT:ASSI:새드 == 0
		RETURN 0
ENDIF
;시간정지중은 스킵
IF FLAG:70 != 1
	;처녀 또한 윤활 부족은 안 돼
	SIF PALAM:윤활 < PALAMLV:2 && TALENT:처녀 && TALENT:처녀 != -1
		RETURN 0
ENDIF

;하반신착의
SIF TEQUIP:0 & 4
	RETURN 0
;목욕탕플레이중은 플레이매트가 없으면 안 돼
IF TEQUIP:31
	SIF ITEM:20 == 0 && NOITEM == 0
		RETURN 0
ENDIF
;후배위로부터는 안 돼
SIF TEQUIP:체위 == 5
	RETURN 0

RETURN 1

path: ERB/커맨드관련/COMF/COMF60 정상위_t.ERB


path: ERB/커맨드관련/COMF/COMF61 후배위_t.ERB


path: ERB/커맨드관련/COMF/COMF62 정상위애널_t.ERB


path: ERB/커맨드관련/COMF/COMF63 후배위애널_t.ERB


path: ERB/커맨드관련/COMF/COMF64 역강간_t.ERB


path: ERB/커맨드관련/COMF/COMF65 기승위_t.ERB


path: ERB/커맨드관련/COMF/COMF66 기승위애널_t.ERB


path: ERB/커맨드관련/COMF/COMF67 대면좌위_t.ERB


path: ERB/커맨드관련/COMF/COMF68 배면좌위_t.ERB


path: ERB/커맨드관련/COMF/COMF69 대면좌위애널_t.ERB


path: ERB/커맨드관련/COMF/COMF7 유두자극_t.ERB


path: ERB/커맨드관련/COMF/COMF70 배면좌위애널_t.ERB


code:
;-------------------------------------------------
;양구멍삽입
;섹스계 파생 커맨드
;-------------------------------------------------
@COM71
VARSET LOCAL
IF GROUPMATCH(TALENT:TARGET:처녀, 1, -1)
	CALL LOST_VIRGIN_STOP(TARGET)
	SIF RESULT
		RETURN 0
ENDIF
SIF SHIRAHU(TARGET)
	CALL 피임요구(TARGET)
IF TFLAG:16 && !(TEQUIP:PLAYER:콘돔 || TCVAR:피임약)
	TFLAG:16 -= 1
	RETURN 0
ENDIF

LOCAL:1 = MASTER_POSE(SET_V, SET_P, 1)
LOCAL:2 = MASTER_POSE(SET_A, SET_P, 1)
IF LOCAL:1
	LOCAL = LOCAL:1
ELSEIF LOCAL:2
	LOCAL = LOCAL:2
ENDIF
;LOCAL 두 구멍삽입 물러날 수 있는 사람
;-------------------------------------------------
;COM별 FLAG의 설정
;-------------------------------------------------
;행위자(이것 설정하는 것은 무리, 잠정적으로 했지만
TCVAR:116 = LOCAL
TCVAR:LOCAL:116 = PLAYER
;-------------------------------------------------
;삽입 부위 관리
;-------------------------------------------------
;PLAYER가 A, TARGET가 V삽입
IF LOCAL == LOCAL:2
	CSTR:10 += @"{LOCAL}/"
	;TCVAR:2 사정장소 플래그(-1=콘돔 1=질내 2=애널 3=수음 4=펠라티오 5=파이즈리 6=가랑이 7=풋잡 8=체표 9=애널 봉사 10=도구 11=촉수 12=A고문 13=V 꾸짖고)
	TCVAR:2 = 1
	;강도
	TCVAR:4 = 1500

	CSTR:PLAYER:10 += @"{LOCAL}/"
	;TCVAR:2 사정장소 플래그(-1=콘돔 1=질내 2=애널 3=수음 4=펠라티오 5=파이즈리 6=가랑이 7=풋잡 8=체표 9=애널 봉사 10=도구 11=촉수 12=A고문 13=V 꾸짖고)
	TCVAR:PLAYER:2 = 2
	;이미 A에 삽입중이니까 이것 필요한 구 없어?
	TCVAR:PLAYER:동정상실 = 1
	TCVAR:A상실발생 = 1
	;강도
	TCVAR:PLAYER:4 = 2500
;PLAYER가 V, TARGET가 A삽입
ELSE
	CSTR:PLAYER:10 += @"{LOCAL}/"
	;TCVAR:2 사정장소 플래그(-1=콘돔 1=질내 2=애널 3=수음 4=펠라티오 5=파이즈리 6=가랑이 7=풋잡 8=체표 9=애널 봉사 10=도구 11=촉수 12=A고문 13=V 꾸짖고)
	TCVAR:PLAYER:2 = 1
	;강도
	TCVAR:PLAYER:4 = 1500

	CSTR:10 += @"{LOCAL}/"
	;TCVAR:2 사정장소 플래그(-1=콘돔 1=질내 2=애널 3=수음 4=펠라티오 5=파이즈리 6=가랑이 7=풋잡 8=체표 9=애널 봉사 10=도구 11=촉수 12=A고문 13=V 꾸짖고)
	TCVAR:2 = 2
	TCVAR:TARGET:동정상실 = 1
	TCVAR:LOCAL:A상실발생 = 1
	;강도
	TCVAR:4 = 2500
ENDIF
;-------------------------------------------------
;소스 추가 처리
;-------------------------------------------------
;소비 체력, 기력
IF FLAG:70 == 1
	BASE:MASTER:TSP -= 220
	SOURCE:PLAYER:쾌C = 400
	SOURCE:쾌C = 400
	SOURCE:성행동= 40
	SOURCE:LOCAL:쾌V = 400
	SOURCE:LOCAL:쾌A = 400
	SOURCE:LOCAL:고통 = 650

	;처녀였던 경우는, 고통의 소스와 반발의 소스를 추가한다
	IF TALENT:LOCAL:처녀 == 1
		SOURCE:LOCAL:고통 += 2750
	ENDIF

	;재생 처녀였던 경우는, 고통의 소스와 반발의 소스를 추가한다
	IF TALENT:LOCAL:처녀 == 2
		SOURCE:LOCAL:고통 += 1750
	ENDIF

	EXP:PLAYER:시간경험 += 1

ELSE

	DOWNBASE:체력 += 40
	DOWNBASE:기력 += 220
	SOURCE:PLAYER:쾌C = 400

	SOURCE:쾌C = 400
	SOURCE:성행동 = 500
	SOURCE:굴종 = 1000
	SOURCE:일탈 = 500
	SOURCE:반감 = 500

	DOWNBASE:LOCAL:체력 += 160
	DOWNBASE:LOCAL:기력 += 350

	SOURCE:LOCAL:쾌V = 400
	SOURCE:LOCAL:쾌A = 400
	SOURCE:LOCAL:사랑 = 300
	SOURCE:LOCAL:성행동 = 300
	SOURCE:LOCAL:고통 = 1300
	SOURCE:LOCAL:굴종 = 500
	SOURCE:LOCAL:일탈 = 500
	SOURCE:LOCAL:반감 = 500

	;-------------------------------------------------
	;특수소스처리
	;-------------------------------------------------
	;억압, 저항 소유에는, 억울소스 추가
	SIF TALENT:자기애 < 0 || TALENT:저항
		SOURCE:울굴 += 500
	SIF TALENT:LOCAL:자기애 < 0 || TALENT:LOCAL:저항
		SOURCE:LOCAL:울굴 += 500

	;정조 관념 소유로 연모, 친애 가져 이외에는 항상 반발의 소스를 추가한다(섹스계만)
	SIF TALENT:LOCAL:정조 && !TALENT:LOCAL:연모
		SOURCE:LOCAL:반감 += 1000

	;처녀였던 경우는, 고통의 소스와 반발의 소스를 추가한다
	IF TALENT:LOCAL:처녀 == 1
		SOURCE:LOCAL:고통 += 5500
		SOURCE:LOCAL:반감 += 3000
	ENDIF

	;재생 처녀였던 경우는, 고통의 소스와 반발의 소스를 추가한다
	IF TALENT:LOCAL:처녀 == 2
		SOURCE:LOCAL:고통 += 3500
		SOURCE:LOCAL:반감 += 1000
	ENDIF
ENDIF
;V에의 고통은 먼저 처리해 둔다
CALL PAIN_CHECK_V(LOCAL)
;A에의 고통은 먼저 처리해 둔다
CALL PAIN_CHECK_A(LOCAL)

;-------------------------------------------------
;더러움의 처리
;-------------------------------------------------
;PLAYER가 V, TARGET가 A삽입
IF LOCAL == LOCAL:1
	;질내에 파과의 피의 더러움을 부가
	SIF TALENT:LOCAL:처녀 && TALENT:처녀 != -1
		STAIN:LOCAL:6 |= 64
	;LOCAL의 질내⇔주인의 P의 더러움이 이동
	IF TALENT:PLAYER:2 & 2
		STAIN:PLAYER:2 |= STAIN:LOCAL:6
		STAIN:PLAYER:2 |= STAIN:LOCAL:3
		STAIN:LOCAL:6 |= STAIN:PLAYER:2
		STAIN:LOCAL:3 |= STAIN:PLAYER:2
	ENDIF
	;LOCAL의 A⇔TARGET의 P의 더러움이 이동
	IF HAS_PENIS(TARGET)
		STAIN:2 |= STAIN:LOCAL:7
		STAIN:2 |= STAIN:LOCAL:4
		STAIN:LOCAL:7 |= STAIN:2
		STAIN:LOCAL:4 |= STAIN:2
	ENDIF
ELSE
	;질내에 파과의 피의 더러움을 부가
	SIF TALENT:LOCAL:처녀 && TALENT:처녀 != -1
		STAIN:LOCAL:6 |= 64
	;LOCAL의 질내⇔TARGET의 P의 더러움이 이동
	IF TALENT:PLAYER:2 & 2
		STAIN:2 |= STAIN:LOCAL:6
		STAIN:2 |= STAIN:LOCAL:3
		STAIN:LOCAL:6 |= STAIN:2
		STAIN:LOCAL:3 |= STAIN:2
	ENDIF
	;LOCAL의 A⇔PLAYER의 P의 더러움이 이동
	IF HAS_PENIS(TARGET)
		STAIN:PLAYER:2 |= STAIN:LOCAL:7
		STAIN:PLAYER:2 |= STAIN:LOCAL:4
		STAIN:LOCAL:7 |= STAIN:PLAYER:2
		STAIN:LOCAL:4 |= STAIN:PLAYER:2
	ENDIF
ENDIF
;-------------------------------------------------
;경험의 처리
;-------------------------------------------------
EXP:PLAYER:삽입경험 ++

;EXP:LOCAL:A경험 ++
EXP:LOCAL:V성교경험 ++
EXP:LOCAL:A성교경험 ++

EXP:삽입경험 ++

;이상경험
IF TALENT:LOCAL:처녀  && TALENT:처녀 != -1 && !GETBIT(CFLAG:3, 9)
	EXP:LOCAL:이상경험 += 2
	SETBIT CFLAG:LOCAL:3, 9
ENDIF

IF LOCAL == TARGET
	IF FLAG:70 == 1
		IF HETEROSEX(LOCAL, PLAYER)
			EXP:PLAYER:(50 + HETEROSEX(TARGET, PLAYER)) += 3
		ENDIF
		EXP:PLAYER:시간경험 += 1
	ELSE
		;동성애 레즈비언이라면 반환값 1 동성애자라면 반환값 2
		IF HETEROSEX(LOCAL, PLAYER)
			EXP:PLAYER:(50 + HETEROSEX(TARGET, PLAYER)) += 3
			EXP:LOCAL:(50 + HETEROSEX(TARGET, PLAYER)) += 3
		ENDIF
	ENDIF
ELSE
	;동성애 레즈비언이라면 반환값 1 동성애자라면 반환값 2
	IF FLAG:70 == 1
		IF HETEROSEX(TARGET, PLAYER)
			EXP:PLAYER:(50 + HETEROSEX(TARGET, PLAYER)) += 3
		ENDIF
		IF HETEROSEX(LOCAL, PLAYER)
			EXP:PLAYER:(50 + HETEROSEX(TARGET, PLAYER)) += 3
		ENDIF
		EXP:PLAYER:시간경험 += 1
	ELSE
		IF HETEROSEX(TARGET, PLAYER)
			EXP:(50 + HETEROSEX(TARGET, PLAYER)) += 3
			EXP:PLAYER:(50 + HETEROSEX(TARGET, PLAYER)) += 3
		ENDIF
		;동성애 레즈비언이라면 반환값 1 동성애자라면 반환값 2
		IF HETEROSEX(LOCAL, PLAYER)
			EXP:PLAYER:(50 + HETEROSEX(TARGET, PLAYER)) += 3
			EXP:LOCAL:(50 + HETEROSEX(TARGET, PLAYER)) += 3
		ENDIF
		IF HETEROSEX(TARGET, LOCAL)
			EXP:(50 + HETEROSEX(TARGET, LOCAL)) += 3
			EXP:LOCAL:(50 + HETEROSEX(TARGET, LOCAL)) += 3
		ENDIF
	ENDIF
ENDIF
RETURN 1

;-------------------------------------------------
;실행 판정
;-------------------------------------------------
@COM_ABLE71
;양구멍삽입 실행 판정
VARSET LOCAL
LOCAL = MASTER_POSE(SET_V, SET_P, 1)
LOCAL:1 = MASTER_POSE(SET_A, SET_P, 1)
IF LOCAL
	LOCAL:2 = LOCAL
ELSEIF LOCAL:1
	LOCAL:2 = LOCAL:1
ENDIF
SIF !TFLAG:100
	RETURN 0
SIF GLOBAL_COMABLE(71)
	RETURN RESULT
[SKIPSTART]
;바이브 삽입중은 안 돼
SIF TEQUIP:13
	RETURN 0
;애널바이브 삽입중은 안 돼
SIF TEQUIP:14
	RETURN 0
;애널비즈 사용중은 불가
SIF TEQUIP:15
	RETURN 0
;보통 관장 사용중은 불가
SIF TEQUIP:21
	RETURN 0
;풍선 사용중은 불가
SIF TEQUIP:22
	RETURN 0
;전극 사용중은 불가
SIF TEQUIP:23
	RETURN 0
[SKIPEND]

;쌍창소유라면 혼자서 가능
IF TALENT:PLAYER:쌍창 == 1
	SIF BASE:PLAYER:발기 < 1000
	RETURN 0
	;자신이 삽입하고 있지 않으면 안 돼
	SIF !LOCAL:2
		RETURN 0
	RETURN 1
ELSE
	;TARGET가 남자, 양성의 경우는 어느 정도 발기하고 있지 않으면 안 돼
	SIF (HAS_PENIS(TARGET)) && BASE:발기 < 1000
		RETURN 0
	;TARGET 이외의누군가에게 삽입하고 있지 않으면 안 돼
	SIF !LOCAL:2 || LOCAL:2 == TARGET
		RETURN 0
	;상대가 처녀로, 순종 4 이하·레즈끼 4 이하의 경우는 할 수 없다(새드라면 OK)
	IF LOCAL:2 && TALENT:(LOCAL:2):처녀
		SIF (ABL:순종 <= 4 || ABL:레즈끼 <= 4) && TALENT:새드 == 0
			RETURN 0
	ENDIF
	;윤활 부족해, 순종 3 이하·레즈끼 3 이하의 ASSI의 경우는 할 수 없다(새드라면 OK)
	IF LOCAL:2 && PALAM:(LOCAL:2):윤활 < PALAMLV:2
		SIF (ABL:순종 <= 3 || ABL:레즈끼 <= 3) && TALENT:새드 == 0
			RETURN 0
	ENDIF
ENDIF
;남자라면 안 돼
SIF LOCAL:2 && !(TALENT:(LOCAL:2):2 & 1)
	RETURN 0
;페니스가 없으면 안 돼
SIF !(HAS_PENIS(TARGET)) && ITEM:PBAND == 0
	RETURN 0
;바이브 장착중은 안 돼
SIF TEQUIP:(LOCAL:2):13
	RETURN 0
;애널바이브 사용중은 불가
SIF TEQUIP:(LOCAL:2):14
	RETURN 0
;애널비즈 사용중은 불가
SIF TEQUIP:(LOCAL:2):15
	RETURN 0
;보통 관장 사용중은 불가
SIF TEQUIP:(LOCAL:2):21
	RETURN 0
;풍선 사용중은 불가
SIF TEQUIP:(LOCAL:2):22
	RETURN 0
;전극 사용중은 불가
SIF TEQUIP:(LOCAL:2):23
	RETURN 0
;새댁플레이중은 안 돼
SIF TEQUIP:(LOCAL:2):33 || TEQUIP:33
	RETURN 0
;체격 차이가 있는 경우는 경험이 없으면 안 돼
SIF 삽입불가(TARGET) == 1 && EXP:A경험 < 50 && !TALENT:PLAYER:금단의지식
	RETURN 0
;한층 더 체격 차이가 있으면(자) 금단의지식이 필요
SIF 삽입불가(TARGET) == 2 && (!TALENT:PLAYER:금단의지식 || EXP:A경험 < 50)
	RETURN 0
SIF 삽입불가(TARGET) == 3
	RETURN 0
;시간정지중은 스킵
IF FLAG:70 != 1
	;일정한 A경험이 필요
	SIF EXP:(LOCAL:2):2 < EXPLV:3
		RETURN 0
ENDIF
;하반신착의
SIF TEQUIP:(LOCAL:2):0 > 1
	RETURN 0

RETURN 1

path: ERB/커맨드관련/COMF/COMF71 양구멍삽입_t.ERB


path: ERB/커맨드관련/COMF/COMF79 난쟁이오나홀_t.ERB


path: ERB/커맨드관련/COMF/COMF8 보지벌리기_t.ERB


path: ERB/커맨드관련/COMF/COMF80 수음_t.ERB


path: ERB/커맨드관련/COMF/COMF81 펠라치오_t.ERB


path: ERB/커맨드관련/COMF/COMF82 파이즈리_t.ERB


path: ERB/커맨드관련/COMF/COMF83 스타마_t.ERB


path: ERB/커맨드관련/COMF/COMF84 거품춤_t.ERB


path: ERB/커맨드관련/COMF/COMF85 풋잡_t.ERB


code:
;빠져 나간다
@COM850
#DIM 달래고판정
#DIM 빠져나갈수없다
달래고판정 = 0
빠져나갈수없다 = 0
IF TCVAR:마음대로해
	PRINTFORMW %마스터는% 바동바동 발버둥 쳤다
	IF RAND:100 < RAND:(50 + ABL:새드끼 * 20 + ABL:전투능력 * 20)
		PRINTFORMW 그러나 헛수고에 끝났다
		CALL KOJO_MESSAGE_SEND("EVENT", 10, TARGET, 3,1)
		DOWNBASE:MASTER:기력 += 50
		DOWNBASE:MASTER:체력 += 50
		SOURCE:가학 += 100
		TCVAR:기학플래그 = 1
		TIME += 5
		RETURN 1
	ENDIF
ELSEIF TFLAG:껴안기모드
	PRINTFORML %마스터는% %타겟을% 뿌리쳤다
	;그 외
	IF CFLAG:TARGET:성격경향 == 5 || !FLAG:카운터지문제어
		PRINTFORML %타겟은% 불만스럽게 하고 있다…
	;심약계
	ELSEIF CFLAG:TARGET:성격경향 == 1
		PRINTFORML %타겟은% 조금 슬픈 듯이 하고 있다…
	;강경세계
	ELSEIF CFLAG:TARGET:성격경향 == 2
		PRINTFORML %타겟은% 무정하네요라고 말해 웃었다…
	;솔직함계
	ELSEIF CFLAG:TARGET:성격경향 == 3
		PRINTFORML %타겟은% %CALLNAME:MASTER%의 반응을 즐기고 있는 것 같다…
	;진면목계
	ELSEIF CFLAG:TARGET:성격경향 == 4
		PRINTFORML %타겟은% %CALLNAME:MASTER%에, 본궤도에 올라 미안하다고 사과했다…
	ENDIF
ELSE
	FOR LOCAL, 1, CHARANUM
		IF CFLAG:LOCAL:현재위치 == CFLAG:MASTER:현재위치 && CFLAG:LOCAL:욕구불만도 >= 500
			달래고판정 = LOCAL
			BREAK
		ENDIF
	NEXT
	IF !달래고판정
		PRINTFORML %마스터는% %CALLNAME:TARGET%\@ GET_TARGETNUM() > 1 ?  들 # \@을 달래 빠져 나갔다
	ELSE
		IF (TALENT:MASTER:성별 & 1) && (TALENT:성별 & 1) && TIME_PROGRESS(TFLAG:덮치기개시시간) >= 360
			PRINTFORMW %CALLNAME:달래고판정%는 조금 물건 충분하고 없을 것 같게 하면서도 %마스터를% 해방했다
			TFLAG:껴안기모드 = 0
			TCVAR:안아다하고금지시간 = 70 + MAX(10 - ABL:욕망, 0) * 3 - MAX((CFLAG:적극성 + CFLAG:수줍음) * 5, 30)
		ELSEIF !TALENT:MASTER:비동정 && TALENT:MASTER:2 & 2
			PRINTFORMW %CALLNAME:달래고판정%에 사정사정해졌다
			PRINTFORMW %CALLNAME:달래고판정%는 아무래도 %CALLNAME:MASTER%의 처음을 갖고 싶은 것 같다
			CFLAG:달래고판정:욕구불만도 += 50
			BASE:달래고판정:기력 -= 200
			TIME += 5
			RETURN 1
		ELSEIF TALENT:MASTER:처녀 || TALENT:달래고판정:처녀
			PRINTFORMW %CALLNAME:달래고판정%는 어색한 듯이 %마스터를% 해방했다
			BASE:달래고판정:기력 -= 100
			BASE:달래고판정:무드 = 0
		ELSE
			PRINTFORMW %CALLNAME:달래고판정%는 아직%마스터를% 해방할 생각은 없는 것 같다
			CALL KOJO_MESSAGE_SEND("EVENT", 10, 달래고판정, 3,2)
			DOWNBASE:MASTER:기력 += 100
			DOWNBASE:MASTER:체력 += 50
			TIME += 5
			RETURN 1
		ENDIF
	ENDIF
ENDIF

CALL 덮쳐지기종료

@덮쳐지기종료(ARGS)
FOR LOCAL, 0, CHARANUM
	IF CFLAG:LOCAL:우후후 == 2 || LOCAL == TARGET
		IF ARGS == "만족"
			CALL KOJO_MESSAGE_SEND("EVENT", 30, LOCAL, 2)
		ELSEIF ARGS == "지쳤다"
			CALL KOJO_MESSAGE_SEND("EVENT", 30, LOCAL, 1)
		;밀어넘어뜨려지고 모드 종료
		ELSEIF CFLAG:LOCAL:우후후 == 2 && LOCAL != 0 && !TFLAG:껴안기모드 && CFLAG:MASTER:현재위치 == CFLAG:LOCAL:현재위치
			CALL KOJO_MESSAGE_SEND("EVENT", 10, LOCAL, 2,0)
		;껴안기모드 만족 종료
		ELSEIF TFLAG:껴안기모드 && TARGET == LOCAL
			CALL KOJO_MESSAGE_SEND("EVENT", 10, LOCAL, 2,1)
		;껴안기모드 강제 종료
		ELSEIF TFLAG:102 == 1 && TARGET == LOCAL
			CALL KOJO_MESSAGE_SEND("EVENT", 10, LOCAL, 2,2)
		ENDIF
		SIF CFLAG:LOCAL:우후후 == 2
			CALL SET_DERAY(5, LOCAL)
		BASE:LOCAL:무드 = BASE:LOCAL:무드 / 2
		BASE:LOCAL:이성 = MIN(BASE:LOCAL:이성 + 300, MAXBASE:LOCAL:이성)
		CFLAG:LOCAL:우후후 = 0
		;우후후시 계 장비 리셋
		CALL ENDUFUFU(LOCAL)
	ENDIF
NEXT
TFLAG:102 = 1
TFLAG:106 = 0
;MASTER와의 접촉 위치를 리셋
CALL TOUCH_SET(0,0,0,1)
[SKIPSTART]
;(은)는만 플래그를 오프
TEQUIP:2 = 0
TEQUIP:상반신은만상태 = 0
;캐릭터 전원의 복재세팅
FOR LOCAL, 0, CHARANUM
	CALL CLOTHES_SETTING_TRAIN(LOCAL)
	;윗도리탈의가끝난상태 플래그를 오프
	TEQUIP:LOCAL:5 = 0
NEXT
[SKIPEND]

TIME += 5
RETURN 1

path: ERB/커맨드관련/COMF/COMF850 빠져나간다_t.ERB


path: ERB/커맨드관련/COMF/COMF88 진공펠라_t.ERB


path: ERB/커맨드관련/COMF/COMF89 콘돔정액먹기_t.ERB


path: ERB/커맨드관련/COMF/COMF9 자위_t.ERB


path: ERB/커맨드관련/COMF/COMF90 애널애무된다_t.ERB


path: ERB/커맨드관련/COMF/COMF91 애널핥아진다_t.ERB


path: ERB/커맨드관련/COMF/COMF92 Ａ정상위된다_t.ERB


path: ERB/커맨드관련/COMF/COMF93 Ａ후배위된다_t.ERB


path: ERB/커맨드관련/COMF/COMF94 Ａ기승위된다_t.ERB


path: ERB/커맨드관련/COMF/COMF95 유사삽입_t.ERB


code:
@붉은실발동
IF TALENT:MASTER:붉은실&&(RAND:20==0)
	SETCOLOR C_RED
	PRINTFORML %CALLNAME:MASTER%의 운명의 붉은 실이 빛나고 있습니다.
	SETCOLOR C_PINK
	PRINTFORML %타겟은% %CALLNAME:MASTER%에게서 운명을 느낀 듯하다.
	RESETCOLOR
	SIF !RELATION:TARGET:MASTER
		RELATION:TARGET:MASTER = 100
	RELATION:TARGET:MASTER += 20
	SIF !RELATION:MASTER:TARGET
		RELATION:MASTER:TARGET = 100
	RELATION:MASTER:TARGET += 20
ENDIF


code:
;-------------------------------------------------
;손을잡는다
;외출 커맨드
;-------------------------------------------------
@COM603
CALL SUC_FAIL(110, MIN(90 + GET_SUCCESS_RATE() / 5, 109))
SELECTCASE TFLAG:193
	;실패
	CASE -1
		SIF TFLAG:192 == -2
			RETURN 0
	;대성공
	CASE 1
		CALL 붉은실발동
		EXP:MASTER:데이트경험 ++
		EXP:TARGET:데이트경험 ++
	;성공
	CASE 0
ENDSELECT

;선제 발생시는 실패하지 않는다
;SIF TFLAG:102 == 3
;	TFLAG:193 = MAX(TFLAG:193, 0)
DOWNBASE:기력 += 50
DOWNBASE:MASTER:기력 += 50

;고정으로 획득하는 소스
SOURCE:환락 = 150
SOURCE:사랑 = 150
SOURCE:복종 = 100
SOURCE:반감 = 100
SOURCE:수동 = 100

;ABL:친밀을 본다
IF ABL:친밀 <= 3
	SOURCE:환락 += 150 + (ABL:친밀 * 50)
	SOURCE:사랑 += 180 + (ABL:친밀 * 40)
ELSEIF ABL:친밀 <= 5
	SOURCE:환락 += 500 + (ABL:친밀 * 50)
	SOURCE:사랑 += 450 + (ABL:친밀 * 30)
ELSEIF ABL:친밀 <= 8
	SOURCE:환락 += 800 + (ABL:친밀 * 50)
	SOURCE:사랑 += 700 + (ABL:친밀 * 50)
ELSEIF ABL:친밀 <= 11
	SOURCE:환락 += 1000 + (ABL:친밀 * 90)
	SOURCE:사랑 += 900 + (ABL:친밀 * 50)
ELSE
	SOURCE:환락 += 1800 + (ABL:친밀 * 40)
	SOURCE:사랑 += 2000 + (ABL:친밀 * 20)
ENDIF

;ABL:순종을 본다
IF ABL:순종 <= 3
	SOURCE:복종 += 100 + (ABL:순종 * 50)
	SOURCE:수동 += 170 + (ABL:순종 * 40)
ELSEIF ABL:순종 <= 5
	SOURCE:복종 += 300 + (ABL:순종 * 50)
	SOURCE:수동 += 350 + (ABL:순종 * 30)
ELSEIF ABL:순종 <= 8
	SOURCE:복종 += 500 + (ABL:순종 * 50)
	SOURCE:수동 += 600 + (ABL:순종 * 50)
ELSEIF ABL:순종 <= 11
	SOURCE:복종 += 800 + (ABL:순종 * 90)
	SOURCE:수동 += 700 + (ABL:순종 * 50)
ELSE
	SOURCE:복종 += 1500 + (ABL:순종 * 40)
	SOURCE:수동 += 1000 + (ABL:순종 * 20)
ENDIF

;호감 도를 본다
IF CFLAG:2 <= 500
	SOURCE:사랑 += CFLAG:2
ELSEIF CFLAG:2 <= 5000
	SOURCE:사랑 += 200 + (CFLAG:2 - 500) / 3
ELSE
	SOURCE:사랑 += 1000 + (CFLAG:2 - 5000) / 5
ENDIF
SIF SOURCE:환락 < 0
	SOURCE:환락 = 0

LOCAL = 0
;친밀이 낮으면 호감도 저하
IF ABL:친밀 == 0
	LOCAL -= 3
ELSEIF ABL:친밀 == 1
	LOCAL -= 2
ELSEIF ABL:친밀 == 2
	LOCAL -= 1
ENDIF
;순종이 낮으면 호감도 저하
IF ABL:순종 == 0
	LOCAL -= 5
ELSEIF ABL:순종 == 1
	LOCAL -= 3
ELSEIF ABL:순종 == 2
	LOCAL -= 1
ENDIF
;호감도가 낮으면 호감도 저하
IF CFLAG:2 <= 50
	LOCAL -= 3
ELSEIF CFLAG:2 <= 100
	LOCAL -= 2
ELSEIF CFLAG:2 <= 250
	LOCAL -= 1
ENDIF

SOURCE:욕정 = 50 + 150 * ABL:욕망
SOURCE:굴종 = 100 + 200 * ABL:마조끼

;호감도 변화
TFLAG:97 += LOCAL

IF TFLAG:193 == -1
	TIMES SOURCE:환락 , 0.10
	TIMES SOURCE:사랑 , 0.50
	TIMES SOURCE:복종 , 0.50
	TIMES SOURCE:수동 , 1.50
	TIMES SOURCE:욕정 , 0.50
	TIMES SOURCE:반감 , 1.50
ELSEIF TFLAG:193 == 0
	TIMES SOURCE:환락 , 1.00
	TIMES SOURCE:사랑 , 1.00
	TIMES SOURCE:복종 , 1.00
	TIMES SOURCE:수동 , 1.00
	TIMES SOURCE:욕정 , 1.00
ELSE
	TIMES SOURCE:환락 , 2.00
	TIMES SOURCE:사랑 , 2.00
	TIMES SOURCE:복종 , 1.50
	TIMES SOURCE:수동 , 2.00
	TIMES SOURCE:욕정 , 2.00
	TIMES SOURCE:반감 , 0.75
ENDIF
TIME += 10
CALL DATE_GOES_ON(1)
CALL 붉은실발동
EXP:MASTER:데이트경험 ++
EXP:TARGET:데이트경험 ++

RETURN 1

;-------------------------------------------------
;실행 가능 판정
;-------------------------------------------------
;손을잡는다
@COM_ABLE603
;일괄관리
SIF GLOBAL_COMABLE(603)
	RETURN RESULT
;상대가 없다
SIF !TARGET
	RETURN 0
;데이트 도중 이외
SIF (CFLAG:현재위치 % 100) % 10 == 0
	RETURN 0
SIF CFLAG:MASTER:데이트중 == MAIN_MAP
	RETURN 0
;초대방
SIF CFLAG:MASTER:현재위치 == OMANEKIBEYA()
	RETURN 0
;전회 손을 잡았다
;SIF PREVCOM == 603
;	RETURN 0
;수면중
SIF CFLAG:수면
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 1

path: ERB/커맨드관련/COMF/외출계/COMF603 손을잡는다_t.ERB


code:
;-------------------------------------------------
;산책한다
;외출 커맨드
;-------------------------------------------------
@COM604
#DIM 현재지점
#DIM 이동처지점
#DIM 소요시간, 11
#DIM 행선지
#DIM 특수쇼트컷
VARSET LOCAL
DRAWLINEFORM ~

CALL DRAW_MAP("ODEKAKE")

DRAWLINEFORM ~
PRINTL 어디로 향합니까?
LOCAL = 6000 + (CFLAG:MASTER:현재위치 / 100) * 10
현재지점 = (CFLAG:MASTER:현재위치 - (CFLAG:MASTER:데이트중 * 100)) / 10
FOR LOCAL:2, 1, 10
	LOCAL ++
	;현재위치는 스킵
	IF (LOCAL - 6000) * 10 == CFLAG:MASTER:현재위치
		PRINT [0] -
		PRINTFORM %NAME_FROM_PLACE(LOCAL), 18, LEFT%
		SETCOLOR 0 , 255, 0
		PRINT 현재지
		RESETCOLOR
		LOCAL:3 ++
		SIF !(LOCAL:3 % 2)
			PRINTL
		CONTINUE
	ENDIF
	;지점명 미정도리가 되면 스킵
	SIF NAME_FROM_PLACE(LOCAL) == "" || LOCAL > 6000 + ((CFLAG:MASTER:현재위치 / 100) * 10) + 9
		CONTINUE
	소요시간:(LOCAL:2) = ABS(현재지점 - LOCAL:2) * 3
	;영원정내부
	SIF LOCAL == 6048
		소요시간:(LOCAL:2) = ABS(현재지점 - 5) * 3 + 3
	;몽환관
	SIF LOCAL == 6049
		소요시간:(LOCAL:2) = ABS(현재지점 - 7) * 3 + 3
	특수쇼트컷 = 0
	;코마치의 능력
	IF FLAG:데이트상대 == [[코마치]]
		소요시간:(LOCAL:2) = (소요시간:(LOCAL:2) * 2) / 3
		특수쇼트컷 = 1
	ENDIF
	;와카사기히메의 부담
	SIF FLAG:데이트상대 == [[와카사기히메]] && !GETBIT (FLAG:풀장, 0)
		소요시간:(LOCAL:2) = (소요시간:(LOCAL:2) * 3) / 2
	;쇼트 컷 효과
	SELECTCASE LOCAL
		;미혹의 죽림
		CASE 6041 TO 6047
			IF SHORTCUT(6)
				소요시간:(LOCAL:2) = (소요시간:(LOCAL:2) * 2) / 3
				특수쇼트컷 = 2
			ENDIF
		;천계
		CASE 6081 TO 6085
			IF CFLAG:MASTER:현재위치 == 천계 && SHORTCUT(2)
				소요시간:(LOCAL:2) = 소요시간:(LOCAL:2) / 2
				특수쇼트컷 = 2
			ENDIF
		CASE 6089
			IF SHORTCUT(2)
				소요시간:(LOCAL:2) = 소요시간:(LOCAL:2) / 2
				특수쇼트컷 = 2
			ENDIF
		;명계
		CASE 6061, 6062
			IF CFLAG:MASTER:현재위치 >= 670 && CFLAG:MASTER:현재위치 <= 690 && SHORTCUT(3)
				소요시간:(LOCAL:2) = 소요시간:(LOCAL:2) / 2
				특수쇼트컷 = 2
			ENDIF
		CASE 6067, 6068, 6069
			IF CFLAG:MASTER:현재위치 >= 610 && CFLAG:MASTER:현재위치 <= 620 && SHORTCUT(3)
				소요시간:(LOCAL:2) = 소요시간:(LOCAL:2) / 2
				특수쇼트컷 = 2
			ENDIF
	ENDSELECT
	PRINTFORM [{LOCAL:2}] -
	PRINTFORM %NAME_FROM_PLACE(LOCAL)%
	PRINTFORM %" " *(20 - STRLENS(NAME_FROM_PLACE(LOCAL)))%
	SIF 소요시간:(LOCAL:2) < 10
		PRINT
	IF 특수쇼트컷 == 1
		SETCOLOR 250, 220, 180
	ELSEIF 특수쇼트컷 == 2
		SETCOLOR 100, 250, 140
	ENDIF
	PRINTFORM {소요시간:(LOCAL:2)}분
	RESETCOLOR
	LOCAL:3 ++
	SIF !(LOCAL:3 % 2)
		PRINTL
NEXT
SIF GROUPMATCH(CFLAG:MASTER:데이트중, 2, 4, 5, 6, 8, 9)
	PRINTL
IF CFLAG:MASTER:현재위치 == 천계 && (ABL:MASTER:교양 >= 2 || ABL:(FLAG:데이트상대):교양 >= 2)
	소요시간 = 10
	SIF SHORTCUT(2)
		소요시간 = 5
	PRINT [67] - 피안
	PRINTFORML {소요시간}분
ELSEIF CFLAG:MASTER:현재위치 == 무연총 && (ABL:MASTER:교양 >= 2 || ABL:(FLAG:데이트상대):교양 >= 2)
	소요시간 = 5
	SIF SHORTCUT(3)
		소요시간 = 3
	PRINT [68] - 명계
	PRINTFORML {소요시간}분
ENDIF
PRINTL [100] - 돌아간다　　　　　　　　　[99] - 다른 지역으로 이동
$INPUT_LOOP
INPUT
행선지 = RESULT * 10 + CFLAG:MASTER:데이트중 * 100
SIF RESULT == 100 || RESULT == 0
	RETURN -1
IF RESULT == 99
	DRAWLINE
	CALL OTHERREGIONS
	IF RESULT == 100
		RETURN -1
	ELSE
	ENDIF
	RETURN 1
ELSEIF RESULT == 67
	IF CFLAG:MASTER:현재위치 == 천계 && (ABL:MASTER:교양 >= 2 || ABL:(FLAG:데이트상대):교양 >= 2)
		CFLAG:MASTER:현재위치 = 피안
		CFLAG:MASTER:데이트중 = 6
		CFLAG:(FLAG:데이트상대):현재위치 = 피안
		CFLAG:(FLAG:데이트상대):데이트중 = 6
		IF FLAG:70
			BASE:MASTER:TSP -= 소요시간 * 2
		ELSE
			TIME += 소요시간
		ENDIF
		RETURN 1
	ELSE
		RETURN -1
	ENDIF
ELSEIF RESULT == 68
	IF CFLAG:MASTER:현재위치 == 무연총 && (ABL:MASTER:교양 >= 2 || ABL:(FLAG:데이트상대):교양 >= 2)
		CFLAG:MASTER:현재위치 = 백옥루정원
		CFLAG:MASTER:데이트중 = 6
		CFLAG:(FLAG:데이트상대):현재위치 = 백옥루정원
		CFLAG:(FLAG:데이트상대):데이트중 = 6
		IF FLAG:70
			BASE:MASTER:TSP -= 소요시간 * 2
		ELSE
			TIME += 소요시간
		ENDIF
		RETURN 1
	ELSE
		RETURN -1
	ENDIF
ELSEIF RESULT < 1 || RESULT > (LOCAL - (6000 + (CFLAG:MASTER:현재위치 / 100) * 10) + 1)
	CLEARLINE 1
	RESTART
ELSEIF NAME_FROM_PLACE(RESULT + 6000 + (CFLAG:MASTER:현재위치 / 100) * 10) == ""
	CLEARLINE 1
	RESTART
ENDIF

CALL SANSAKU_UNIQUEPLACE(행선지)
SIF RESULT == -1
	RESTART

이동처지점 = RESULT
행선지 = 이동처지점 * 10 + CFLAG:MASTER:데이트중 * 100
[IF_DEBUG]
PRINTFORML 이동처 지점은 {이동처지점}, 행선지는 {행선지}
[ENDIF]
FOR LOCAL, 0, CHARANUM
	SIF CHK_DATENOW(CFLAG:LOCAL:데이트중)
		CFLAG:LOCAL:현재위치 = 행선지
NEXT
IF FLAG:70 == 1
	BASE:MASTER:TSP -= 소요시간:이동처지점 * 2
	IF TFLAG:운반
		BASE:MASTER:TSP -= 소요시간:이동처지점 / 3
		CFLAG:(TFLAG:운반):현재위치 = CFLAG:MASTER:현재위치
	ENDIF
ELSEIF GETBIT(FLAG:순간이동, 0) && BASE:MASTER:TSP > 소요시간:이동처지점 * 2 && !FLAG:데이트상대
	BASE:MASTER:TSP -= 소요시간:이동처지점 * 2
ELSE
	TIME += 소요시간:이동처지점
	SIF CHK_DATENOW(CFLAG:93:데이트중) && !GETBIT (FLAG:풀장, 0)
		BASE:MASTER:체력 = MAX(BASE:MASTER:체력 - 50,500)
	PRINTFORML {소요시간:이동처지점}분 걸려 %STR:(6000 + 행선지 / 10)%로 이동했다
ENDIF

RETURN 1

;-------------------------------------------------
;실행 가능 판정
;-------------------------------------------------
;산책한다
@COM_ABLE604
SIF TFLAG:망상중
	RETURN 0
;일괄관리
SIF GLOBAL_COMABLE(604)
	RETURN RESULT
;데이트중 이외
SIF AT_HOME(MASTER)
	RETURN 0
SIF CFLAG:우후후
	RETURN 0
;데이트 도중
SIF (CFLAG:현재위치 % 100) % 10
	RETURN 0
;망상중
SIF TFLAG:망상중 == 1
	RETURN 0
RETURN 1

@SANSAKU_UNIQUEPLACE(ARG)
#DIM 현재지점
#DIM 이동처지점

현재지점 = (CFLAG:MASTER:현재위치 - (CFLAG:MASTER:데이트중 * 100)) / 10
이동처지점 = (ARG - (CFLAG:MASTER:데이트중 * 100)) / 10

SELECTCASE ARG
	CASE 50
		IF ABL:MASTER:교양 < 1
			PRINTFORMW 미숙한 %CALLNAME:MASTER%에는 찾아낼 수가 없다…
			RETURN -1
		ENDIF
	CASE 피안, 백옥루정원, 백옥루, 천계
		IF ABL:MASTER:교양 < 1 && ABL:(FLAG:데이트상대):교양 < 1
			PRINTFORMW 좀 더 교양을 붙이지 않으면 위험하다
			CLEARLINE 2
			RETURN -1
		ENDIF
	CASE 달의도시, 와타츠키정, 옥토거리, 복숭아숲～모래사장, 고요의바다
		IF ABL:MASTER:교양 < 2 && ABL:(FLAG:데이트상대):교양 < 2
			PRINTFORMW 좀 더 교양을 붙이지 않으면 위험하다
			CLEARLINE 2
			RETURN -1
		ENDIF
	;세이자 하우스의 전
	CASE 740
		;본인은 들어갈 수 있는;연모가 있으면 들어갈 수 있어 줄 수 있는;전생에서 정사가 연모 or 연인이라면 장소는 안다
		IF ARG:1 == [[세이자]] || TALENT:[[세이자]]:연모 || GETBIT (CFLAG:[[세이자]]:인계, 0) || GETBIT (CFLAG:[[세이자]]:인계, 1)
		ELSE
			PRINTFORMW 어디에 있을지 모른다…
			RETURN -1
		ENDIF
	;카센하우스
	CASE 750
		;카센이 연모;본인은 들어갈 수 있다
		IF TALENT:[[카센]]:연모 || ARG:1 == [[카센]]
		;데이트상대가 텐시나 자신이 텐시
		ELSEIF FLAG:데이트상대 == [[텐시]] || ARG:1 == [[텐시]]
		;교양이 부족하면 가까스로 도착할 수 없다
		ELSEIF ABL:(ARG:1):교양 < 3
			PRINTFORMW 조금 전부터 저택과의 거리가 전혀 줄어들지 않는다……
			PRINTFORMW 같은 곳을 빙빙 돌아 있는 것과 같다
			RETURN -1
		;그 이외는 전투력 3이상 필요
		ELSEIF ABL:(ARG:1):전투능력 < 3 && !FLAG:70
			PRINTFORMW 뜰에 웅크리고 있던 경시하는 범이, %CALLNAME:MASTER%의 모습을 인정해 신음소리를 내 소리를 높였다
			PRINTFORMW 가까워지지 않는 것이 좋은 것 같다
			RETURN -1
		ENDIF
	CASE 480
		IF !FLAG:70 && !영원정침입허가()
			PRINTL 안쪽에 들어가려고 했지만, 입구에서 토끼들에게 제지당했다
			IF 현재지점 == 5
				RETURN -1
			ELSE
				RETURN 5
			ENDIF
		ENDIF
		SIF FLAG:70
			PRINTFORMW 무단으로 들어갔다
	CASE 혼숙여관, 러브호텔
		IF !FLAG:데이트상대
			PRINTW 혼자서 가도 어쩔 수 없다
			CLEARLINE 2
			RETURN -1
		ENDIF
		IF MONEY < 3000
			PRINTW 돈이 부족하다
			CLEARLINE 2
			RETURN -1
		ENDIF
		SIF !GETBIT(CFLAG:(FLAG:데이트상대):기정사실 , 1)
			CALL 껴안고((FLAG:데이트상대))
		IF TCVAR:(FLAG:데이트상대):껴안고첫회 >= 6 && !TCVAR:(FLAG:데이트상대):만취
				PRINTFORMW %CALLNAME:(FLAG:데이트상대)%에 들어가는 것을 거부되어 버렸다
				IF TCVAR:(FLAG:데이트상대):껴안고첫회 == 6
					PRINTFORMW 그러나 %CALLNAME:(FLAG:데이트상대)%는 꼭 싫지만은 않은 것 같다
					CLEARLINE 1
				ENDIF
			CLEARLINE 2
			RETURN -1
		ELSE
			PRINTL \3000걸립니다. 괜찮겠습니까?
			CALL ASK_YN
			IF RESULT == 0
				MONEY -= 3000
				IF TCVAR:(FLAG:데이트상대):만취
					PRINTFORMW %CALLNAME:(FLAG:데이트상대)%가 몽롱해지고 있는 것을 말해 일로, %마스터는% 그녀를 %STR:(6000 + ARG / 10)%로 데리고 들어갔다
				ELSE
					PRINTFORMW %마스터는% %CALLNAME:(FLAG:데이트상대)%과 %STR:(6000 + ARG / 10)%에 들어갔다
				ENDIF
				RETURN 이동처지점
			ELSE
				CFLAG:(FLAG:데이트상대):우후후 = 0
				CFLAG:MASTER:우후후 = 0
				RETURN -1
			ENDIF
			CFLAG:(FLAG:데이트상대):우후후 = 0
			CFLAG:MASTER:우후후 = 0
		ENDIF
ENDSELECT

RETURN 이동처지점

path: ERB/커맨드관련/COMF/외출계/COMF604 산책한다_t.ERB


code:
;-------------------------------------------------
;지나치는길한다
;외출 커맨드
;-------------------------------------------------
@COM605
CALL SUC_FAIL()
SELECTCASE TFLAG:193
	;실패
	CASE -1
		SIF TFLAG:192 == -2
			RETURN 0
	;대성공
	CASE 1
		CALL 붉은실발동
		EXP:MASTER:회화경험 ++
		EXP:MASTER:데이트경험 ++
		EXP:TARGET:데이트경험 ++
	;성공
	CASE 0
ENDSELECT
;선제 발생시는 실패하지 않는다
SIF TCVAR:20
	TFLAG:193 = MAX(TFLAG:193, 0)
DOWNBASE:MASTER:기력 += 50
DOWNBASE:MASTER:체력 += 30
DOWNBASE:기력 += 30
DOWNBASE:체력 += 20

;고정으로 획득하는 소스
SOURCE:환락 = 500
SOURCE:사랑 = 300

;ABL:순종을 본다
IF ABL:순종 <= 1
	SOURCE:환락 += (ABL:순종 * 60)
	SOURCE:사랑 += (ABL:친밀 * 30)
ELSEIF ABL:순종 <= 3
	SOURCE:환락 += 200 + (ABL:순종 * 60)
	SOURCE:사랑 += 200 + (ABL:친밀 * 30)
ELSEIF ABL:순종 <= 5
	SOURCE:환락 += 500 + (ABL:순종 * 60)
	SOURCE:사랑 += 500 + (ABL:친밀 * 30)
ELSEIF ABL:순종 <= 8
	SOURCE:환락 += 750 + (ABL:순종 * 75)
	SOURCE:사랑 += 750 + (ABL:친밀 * 45)
ELSEIF ABL:순종 <= 11
	SOURCE:환락 += 1000 + (ABL:순종 * 75)
	SOURCE:사랑 += 1000 + (ABL:친밀 * 50)
ELSE
	SOURCE:환락 += 2000 + (ABL:순종 * 30)
	SOURCE:사랑 += 2000 + (ABL:친밀 * 20)
ENDIF

;호감 도를 본다
IF CFLAG:2 <= 500
	SOURCE:환락 += CFLAG:2
	SOURCE:사랑 += CFLAG:2
ELSEIF CFLAG:2 <= 5000
	SOURCE:환락 += 500 + (CFLAG:2 - 500) / 3
	SOURCE:사랑 += 300 + (CFLAG:2 - 300) / 3
ELSE
	SOURCE:환락 += 2000 + (CFLAG:2 - 5000) / 5
	SOURCE:사랑 += 1000 + (CFLAG:2 - 1000) / 5
ENDIF
SIF SOURCE:환락 < 0
	SOURCE:환락 = 0
SIF SOURCE:사랑 < 0
	SOURCE:사랑 = 0

SOURCE:수동 = 200 + 100 * ABL:순종
SOURCE:정복 = 200 + 100 * ABL:새드끼

IF TFLAG:193 == -1
	TIMES SOURCE:환락 , 0.10
	TIMES SOURCE:정복 , 0.50
	TIMES SOURCE:수동 , 0.50
ELSEIF TFLAG:193 == 0
	TIMES SOURCE:환락 , 1.00
	TIMES SOURCE:정복 , 1.00
	TIMES SOURCE:수동 , 1.00
ELSE
	TIMES SOURCE:환락 , 2.00
	TIMES SOURCE:정복 , 2.00
	TIMES SOURCE:수동 , 2.00
ENDIF

;지나치는 길 한 플래그
TCVAR:309 = 1440 * DAY + TIME
TIME += 10
CALL 붉은실발동
EXP:MASTER:데이트경험 ++
EXP:TARGET:데이트경험 ++

RETURN 1

;-------------------------------------------------
;실행 가능 판정
;-------------------------------------------------
;지나치는길한다
@COM_ABLE605
;일괄관리
SIF GLOBAL_COMABLE(605)
	RETURN RESULT
;상대가 없다
SIF !TARGET
	RETURN 0
;데이트중 이외
SIF !CHK_DATENOW(CFLAG:12)
	RETURN 0
;데이트 도중 이외
SIF (CFLAG:현재위치 % 100) % 10 == 0
	RETURN 0
;초대방
SIF CFLAG:MASTER:현재위치 == OMANEKIBEYA()
	RETURN 0
;30분간격
SIF TIME_PROGRESS(TCVAR:309) < 30
	RETURN 0
;수면중
SIF CFLAG:수면
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 1

path: ERB/커맨드관련/COMF/외출계/COMF605 어디들른다_t.ERB


code:
;-------------------------------------------------
;식사(데이트)
;외출 커맨드
;-------------------------------------------------
@COM610
#DIM 회복전체력1
#DIM 회복전기력1
#DIM 회복후체력1
#DIM 회복후기력1
#DIM 회복전체력2
#DIM 회복전기력2
#DIM 회복후체력2
#DIM 회복후기력2

회복전체력1 = BASE:MASTER:체력
회복전기력1 = BASE:MASTER:기력
회복전체력2 = BASE:TARGET:체력
회복전기력2 = BASE:TARGET:기력
LOCAL = RAND:100
LOCAL:1 = 90 + GET_SUCCESS_RATE() / 5
SIF LOCAL:1 > 99
	LOCAL:1 = 99
IF LOCAL <= 9
	;10% 확률대성공
		EXP:MASTER:회화경험 ++
		CALL 붉은실발동
		EXP:MASTER:데이트경험 ++
		EXP:TARGET:데이트경험 ++
		TFLAG:193 = 1
ELSEIF LOCAL >= LOCAL:1
	;10～1％で실패
	TFLAG:193 = -1
ELSE
	;残りは성공
	TFLAG:193 = 0
ENDIF
TFLAG:98 += 2

DOWNBASE:기력 += 10

;고정으로 획득하는 소스
SOURCE:환락 = 800
SOURCE:사랑 = 300

;ABL:욕망을 본다
IF ABL:욕망 <= 1
	SOURCE:환락 += (ABL:욕망 * 60)
ELSEIF ABL:욕망 <= 3
	SOURCE:환락 += 300 + (ABL:욕망 * 60)
ELSEIF ABL:욕망 <= 5
	SOURCE:환락 += 600 + (ABL:욕망 * 60)
ELSEIF ABL:욕망 <= 8
	SOURCE:환락 += 900 + (ABL:욕망 * 75)
ELSEIF ABL:욕망 <= 11
	SOURCE:환락 += 1500 + (ABL:욕망 * 75)
ELSE
	SOURCE:환락 += 2500 + (ABL:욕망 * 30)
ENDIF

;ABL:봉사정신을 본다
IF ABL:봉사정신 <= 1
	SOURCE:사랑 += (ABL:봉사정신 * 30)
ELSEIF ABL:봉사정신 <= 3
	SOURCE:사랑 += 200 + (ABL:봉사정신 * 30)
ELSEIF ABL:봉사정신 <= 5
	SOURCE:사랑 += 500 + (ABL:봉사정신 * 30)
ELSEIF ABL:봉사정신 <= 8
	SOURCE:사랑 += 700 + (ABL:봉사정신 * 45)
ELSEIF ABL:봉사정신 <= 11
	SOURCE:사랑 += 1200 + (ABL:봉사정신 * 45)
ELSE
	SOURCE:사랑 += 1800 + (ABL:봉사정신 * 20)
ENDIF

;호감 도를 본다
IF CFLAG:2 <= 500
	SOURCE:환락 += CFLAG:2
ELSEIF CFLAG:2 <= 5000
	SOURCE:환락 += 750 + (CFLAG:2 - 500) / 3
ELSE
	SOURCE:환락 += 2000 + (CFLAG:2 - 5000) / 4
ENDIF
SIF SOURCE:환락 < 0
	SOURCE:환락 = 0

SOURCE:수동 = 200 + 100 * ABL:순종
SOURCE:정복 = 200 + 100 * ABL:새드끼

IF TFLAG:193 == -1
	TIMES SOURCE:환락 , 0.10
	TIMES SOURCE:사랑 , 0.10
	TIMES SOURCE:정복 , 0.50
	TIMES SOURCE:수동 , 0.50
ELSEIF TFLAG:193 == 0
	TIMES SOURCE:환락 , 1.00
	TIMES SOURCE:사랑 , 1.00
	TIMES SOURCE:정복 , 1.00
	TIMES SOURCE:수동 , 1.00
ELSE
	TIMES SOURCE:환락 , 2.00
	TIMES SOURCE:사랑 , 1.50
	TIMES SOURCE:정복 , 2.00
	TIMES SOURCE:수동 , 2.00
ENDIF
SIF TALENT:대식가
	TIMES SOURCE:환락 , 1.50

CALL RECOVER_PERMIL(MASTER, 75,"체력")
CALL RECOVER_PERMIL(MASTER, 150,"기력")
CALL RECOVER_PERMIL(TARGET, 75,"체력")
CALL RECOVER_PERMIL(TARGET, 150,"기력")

TIME += 60
CALL 붉은실발동
EXP:MASTER:데이트경험 ++
EXP:TARGET:데이트경험 ++
CALL 만복도상승(TARGET,"주식")
CALL 만복도상승(MASTER,"주식")
MONEY -= 3000
회복후체력1 = BASE:MASTER:체력
회복후기력1 = BASE:MASTER:기력
회복후체력2 = BASE:TARGET:체력
회복후기력2 = BASE:TARGET:기력
DRAWLINE
PRINTFORML 체력+{ABS(회복후체력1 - 회복전체력1)}(%CALLNAME:MASTER%)
PRINTFORML 기력+{ABS(회복후기력1 - 회복전기력1)}(%CALLNAME:MASTER%)
PRINTFORML 체력+{ABS(회복후체력2 - 회복전체력2)}(%CALLNAME:TARGET%)
PRINTFORML 기력+{ABS(회복후기력2 - 회복전기력2)}(%CALLNAME:TARGET%)
RETURN 1

;-------------------------------------------------
;실행 가능 판정
;-------------------------------------------------
;식사제의를한다
@COM_ABLE610
;일괄관리
SIF GLOBAL_COMABLE(610)
	RETURN RESULT
;상대가 없다
SIF !TARGET
	RETURN 0
;데이트중 이외
SIF !CHK_DATENOW(CFLAG:12)
	RETURN 0
SIF !DATE_PARTY(CFLAG:MASTER:현재위치)
	RETURN 0
SIF CFLAG:행동 == 5
	RETURN 0
;먹었던 바로 직후
SIF !TIME_PROGRESS(TCVAR:312)
	RETURN 0
SIF MONEY < 3000
	RETURN 0
;수면중
SIF CFLAG:수면
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 1

;FileName_DATE_CMN.ERB ----------------------------- Rev1. 00
;데이트처 식사 가부
;CALL		USER
;ARG		현재위치
;RETURN		식사 장소 종류
;COMMENT
;-----------------------------------------------------------
@DATE_PARTY(ARG)
#FUNCTION
IF !AT_HOME(MASTER)
	SELECTCASE ARG
		;특수 식사처
	;	CASE 230,340,350
	;		RETURNF ARG
		;%데이트 지점%의 연회에∼…
	;	CASE 450,650,740,830
	;		RETURNF 2
		;%데이트 지점%로 식사를∼…
		;CASE 상가거리, 죽림입구, 영원정, 백옥루, 옛지옥큰길
		;백옥루 or영원정 데, 데이트 안에 식사는 이상하지 않아?　(이)란 것으로, 가게가 있는 곳(중)에 마음대로 변경
		CASE 상가거리, 죽림입구, 중유의길, 옛지옥큰길
			IF (ARG == 죽림입구 && BASE:28:일량 > 3000 && CFLAG:28:행동 == 5)
				RETURNF 0
			ELSE
				RETURNF 1
			ENDIF
	ENDSELECT
ELSE
	;現在のところは死に分岐
	SELECTCASE ARG
		;지저 술집=옛지옥큰길
		CASE 933
			RETURNF 1
		;-포장마차=죽림입구
		;CASE 444
		;	SIF BASE:28:일량 <= 3000 && CFLAG:28:행동 == 5
		;		RETURNF 1
		;아직 없음=중유의길

		;마을 일품 요리점=상가거리
		CASE 218
			RETURNF 1
	ENDSELECT
ENDIF
RETURNF 0

path: ERB/커맨드관련/COMF/외출계/COMF610 식사（데이트）_t.ERB


code:
@COM613
;돈 벌어 한다
LOCAL = RAND:100
LOCAL:1 = 90 + GET_SUCCESS_RATE() / 5
SIF LOCAL:1 > 99
	LOCAL:1 = 99
IF LOCAL <= 9
	;10%로 대성공
	TFLAG:193 = 1
ELSEIF LOCAL >= LOCAL:1
	;10~1%로 실패
	TFLAG:193 = -1
ELSE
	;나머지는 성공
	TFLAG:193 = 0
ENDIF

DOWNBASE:MASTER:체력 += 50
DOWNBASE:MASTER:기력 += 100

;보수 계산
TFLAG:200 = 0
IF DATE_PARTTIME(CFLAG:MASTER:현재위치) > 99
	TFLAG:200 = (ABL:MASTER:기교 + 10) * (RAND:20 + 40) * 2
ELSE
	TFLAG:200 = (ABL:MASTER:기교 + 10) * (RAND:20 + 30) * DATE_PARTTIME(CFLAG:MASTER:현재위치)
ENDIF
SIF TFLAG:200 > 0
	MONEY += TFLAG:200

IF TARGET > 0 && !CFLAG:수면
	;고정으로 획득하는 소스
	SOURCE:환락 = 100
	SOURCE:반감 = 50

	;ABL:순종을 본다
	IF ABL:순종 <= 1

	ELSEIF ABL:순종 <= 3
		SOURCE:환락 += 100 + (ABL:순종 * 20)
	ELSEIF ABL:순종 <= 5
		SOURCE:환락 += 300 + (ABL:순종 * 30)
	ELSEIF ABL:순종 <= 8
		SOURCE:환락 += 600 + (ABL:순종 * 50)
	ELSEIF ABL:순종 <= 11
		SOURCE:환락 += 1000 + (ABL:순종 * 70)
	ELSE
		SOURCE:환락 += 2000 + (ABL:순종 * 50)
	ENDIF

	;호감 도를 본다
	IF CFLAG:2 <= 500
		SOURCE:환락 += CFLAG:2
	ELSEIF CFLAG:2 <= 5000
		SOURCE:환락 += 500 + (CFLAG:2 - 500) / 3
	ELSE
		SOURCE:환락 += 2000 + (CFLAG:2 - 5000) / 5
	ENDIF
	SIF SOURCE:환락 < 0
		SOURCE:환락 = 0

	SOURCE:정복 = 300 + 100 * ABL:새드끼

ENDIF
TIME += 60

RETURN 1

;-------------------------------------------------
;실행 가능 판정
;-------------------------------------------------
;돈 벌어 한다
@COM_ABLE613
;일괄관리
SIF GLOBAL_COMABLE(613)
	RETURN RESULT
;데이트중 이외
SIF !CHK_DATENOW(CFLAG:MASTER:데이트중)
	RETURN 0
;데이트 도중
SIF (CFLAG:MASTER:현재위치 % 100) % 10
	RETURN 0
;현재위치에 금책 원래가 없다
SIF !DATE_PARTTIME(CFLAG:MASTER:현재위치)
	RETURN 0
;수면중
SIF CFLAG:MASTER:수면
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 0

path: ERB/커맨드관련/COMF/외출계/COMF613 돈벌이한다_t.ERB


code:
;-------------------------------------------------
;식재료의 시장
;외출 커맨드
;-------------------------------------------------
@COM614
#DIM 구입품목
#DIM 가격
#DIM 구입상한
SIF !FLAG:일일이벤트 % 10
	PRINTL 오늘은 특매일 같다(표기 가격에서 50%를 뺍니다)
PRINTFORML 소지금:\\{MONEY}
CALL EX_SHOPPING("잡곡","피나 조. 가난한 사람의 음식", 48)
CALL EX_SHOPPING("묵은쌀","서민의 음식", 50)
CALL EX_SHOPPING("백미","부르주아의 음식", 47)
PRINTL [99] - 그만둔다
INPUT
SIF RESULT == 99
	RETURN -1
구입품목 = RESULT
가격 = ITEMPRICE:구입품목
SIF !FLAG:일일이벤트 % 10
	가격 /= 2

PRINTFORM 몇 개 사겠습니까?  1개　￥{가격}
IF CHK_DATENOW(CFLAG:TARGET:데이트중) && TARGET > 0
	구입상한 = ITEMSALES:구입품목 > 0 ?  MIN( ITEMSALES:구입품목, 10 ) # 10
ELSE
	구입상한 = ITEMSALES:구입품목 > 0 ?  MIN( ITEMSALES:구입품목, 5 ) # 5
ENDIF
PRINTFORML ({구입상한}개까지)
PRINTL [0] - 그만둔다
SIF 구입상한
	PRINT [1] - 1개
SIF 구입상한 >= 5
	PRINT [5] - 5개
SIF 구입상한 >= 10
	PRINT [10] - 10
SIF NOSAMES(구입상한, 0, 1, 5, 10 )
	PRINTFORM [{구입상한}] - {구입상한}개
$LOOP_INPUT
INPUT
IF RESULT < 1
	RETURN -1
ELSEIF RESULT > 5 && !(CHK_DATENOW(CFLAG:TARGET:데이트중) && TARGET > 0)
	PRINTL 혼자서는 그렇게 들 수 없다
	GOTO LOOP_INPUT
ELSEIF RESULT > 10
	PRINTL 그렇게 들 수 없다
	GOTO LOOP_INPUT
ELSEIF MONEY < 가격 * RESULT
	PRINTL 돈이 부족하다
	GOTO LOOP_INPUT
ELSEIF RESULT > 구입상한
	PRINTL 품절입니다
	GOTO LOOP_INPUT
ELSE
	MONEY -= 가격 * RESULT
	ITEM:구입품목 += RESULT
	ITEMSALES:구입품목 -= RESULT
	SIF ITEMSALES:구입품목 == 0
		ITEMSALES:구입품목 = -1	; 제고품절
	SETBIT TFLAG:일일일회, 2
	PRINTFORML {RESULT}개의 %ITEMNAME:구입품목%를 샀다
ENDIF

IF CHK_DATENOW(CFLAG:TARGET:데이트중) && TARGET > 0
	LOCAL = RAND:100
	LOCAL:1 = 90 + GET_SUCCESS_RATE() / 5
	SIF LOCAL:1 > 99
		LOCAL:1 = 99
	IF LOCAL <= 9
		;10%로 대성공
		EXP:MASTER:회화경험 ++
		CALL 붉은실발동
		EXP:MASTER:데이트경험 ++
		EXP:TARGET:데이트경험 ++
		TFLAG:193 = 1
	ELSEIF RESULT == 0
		;사지 않으면 실패
		TFLAG:193 = -1
	ELSE
		;나머지는 성공
		TFLAG:193 = 0
	ENDIF

	DOWNBASE:MASTER:기력 += 50
	DOWNBASE:MASTER:체력 += 30
	DOWNBASE:기력 += 30
	DOWNBASE:체력 += 20

	;고정으로 획득하는 소스
	SOURCE:환락 = 500
	SOURCE:사랑 = 200

	;ABL:순종을 본다
	IF ABL:순종 <= 1
		SOURCE:환락 += (ABL:순종 * 60)
		SOURCE:사랑 += (ABL:친밀 * 30)
	ELSEIF ABL:순종 <= 3
		SOURCE:환락 += 200 + (ABL:순종 * 60)
		SOURCE:사랑 += 150 + (ABL:친밀 * 30)
	ELSEIF ABL:순종 <= 5
		SOURCE:환락 += 400 + (ABL:순종 * 60)
		SOURCE:사랑 += 300 + (ABL:친밀 * 30)
	ELSEIF ABL:순종 <= 8
		SOURCE:환락 += 550 + (ABL:순종 * 75)
		SOURCE:사랑 += 450 + (ABL:친밀 * 45)
	ELSEIF ABL:순종 <= 11
		SOURCE:환락 += 800 + (ABL:순종 * 75)
		SOURCE:사랑 += 600 + (ABL:친밀 * 50)
	ELSE
		SOURCE:환락 += 1500 + (ABL:순종 * 30)
		SOURCE:사랑 += 1200 + (ABL:친밀 * 20)
	ENDIF

	;호감 도를 본다
	IF CFLAG:2 <= 500
		SOURCE:환락 += CFLAG:2
		SOURCE:사랑 += CFLAG:2
	ELSEIF CFLAG:2 <= 5000
		SOURCE:환락 += 500 + (CFLAG:2 - 500) / 3
		SOURCE:사랑 += 300 + (CFLAG:2 - 300) / 3
	ELSE
		SOURCE:환락 += 2000 + (CFLAG:2 - 5000) / 5
		SOURCE:사랑 += 1000 + (CFLAG:2 - 1000) / 5
	ENDIF
	SIF SOURCE:환락 < 0
		SOURCE:환락 = 0
	SIF SOURCE:사랑 < 0
		SOURCE:사랑 = 0

	SOURCE:수동 = 200 + 100 * ABL:순종
	SOURCE:정복 = 200 + 100 * ABL:새드끼

	IF TFLAG:193 == -1
		TIMES SOURCE:환락 , 0.10
		TIMES SOURCE:정복 , 0.50
		TIMES SOURCE:수동 , 0.50
	ELSEIF TFLAG:193 == 0
		TIMES SOURCE:환락 , 1.00
		TIMES SOURCE:정복 , 1.00
		TIMES SOURCE:수동 , 1.00
	ELSE
		TIMES SOURCE:환락 , 2.00
		TIMES SOURCE:정복 , 2.00
		TIMES SOURCE:수동 , 2.00
	ENDIF
	CALL 붉은실발동
	EXP:MASTER:데이트경험 ++
	EXP:TARGET:데이트경험 ++
ENDIF

TIME += 30
RETURN 1

;-------------------------------------------------
;실행 가능 판정
;-------------------------------------------------
;식재료의 시장
@COM_ABLE614
;일괄관리
SIF GLOBAL_COMABLE(614)
	RETURN RESULT
IF !AT_HOME(MASTER)
	SIF CFLAG:MASTER:현재위치 != 상가거리
		RETURN 0
ELSE
	SIF CFLAG:MASTER:현재위치 != 216
		RETURN 0
ENDIF
;이제(벌써) 사고 있어
SIF GETBIT(TFLAG:일일일회, 2)
	RETURN 0
;수면중
SIF CFLAG:MASTER:수면
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 1

path: ERB/커맨드관련/COMF/외출계/COMF614 식재료장보기_t.ERB


code:
;-------------------------------------------------
;도시락을 먹는다
;외출 커맨드
;-------------------------------------------------
@COM615
#DIM 회복전체력1
#DIM 회복전기력1
#DIM 회복후체력1
#DIM 회복후기력1
#DIM 회복전체력2
#DIM 회복전기력2
#DIM 회복후체력2
#DIM 회복후기력2

회복전체력1 = BASE:MASTER:체력
회복전기력1 = BASE:MASTER:기력
회복전체력2 = BASE:TARGET:체력
회복전기력2 = BASE:TARGET:기력
LOCAL = RAND:3
LOCAL:1 = RAND:10
IF LOCAL > ABL:요리기능
	;요리가 서투르다면 실패
	TFLAG:193 = -1
	TFLAG:98 = -1
ELSEIF LOCAL:1 <= ABL:요리기능 - 2
	;요리가 능숙하면 대성공
	EXP:MASTER:회화경험 ++
	CALL 붉은실발동
	EXP:MASTER:데이트경험 ++
	EXP:TARGET:데이트경험 ++
	TFLAG:193 = 1
	TFLAG:98 = 5
ELSE
	;나머지는 성공
	TFLAG:193 = 0
	TFLAG:98 = 3
ENDIF

DOWNBASE:기력 += 50

;고정으로 획득하는 소스
SOURCE:환락 = 300
SOURCE:사랑 = 300
SOURCE:노출 = 100
SOURCE:반감 = 50

IF (ABL:친밀 + ABL:봉사정신) < 10
	LOCAL:2 = 1
	SOURCE:환락 += 1000 + (ABL:친밀 * 100)
	SOURCE:달성 += 500 + (ABL:친밀 * 50)
	SOURCE:사랑 += (ABL:친밀 * 30)
ELSEIF (ABL:친밀 + ABL:봉사정신) < 15
	LOCAL:2 = 2
	SOURCE:환락 += 2000 + (ABL:친밀 * 150)
	SOURCE:달성 += 1500 + (ABL:친밀 * 80)
	SOURCE:사랑 += 400 + (ABL:친밀 * 30)
ELSEIF (ABL:친밀 + ABL:봉사정신) < 25
	LOCAL:2 = 3
	SOURCE:환락 += 3000 + (ABL:친밀 * 200)
	SOURCE:달성 += 2000 + (ABL:친밀 * 100)
	SOURCE:사랑 += 600 + (ABL:친밀 * 40)
ELSE
	 LOCAL:2 = 5
	SOURCE:환락 += 5000 + (ABL:친밀 * 250)
	SOURCE:달성 += 3000 + (ABL:친밀 * 150)
	SOURCE:사랑 += 1800 + (ABL:친밀 * 30)
ENDIF

EXP:요리경험 += LOCAL:2
SIF TALENT:연모
	EXP:애정경험 += LOCAL:2

IF ABL:요리기능 < 2
	CALL RECOVER_PERMIL(TARGET, 40,"체력")
	CALL RECOVER_PERMIL(TARGET, 100,"기력")
	CALL RECOVER_PERMIL(MASTER, 40,"체력")
	CALL RECOVER_PERMIL(MASTER, 100,"기력")
ELSEIF ABL:요리기능 < 4
	BASE:체력 = MIN(BASE:체력 + MAXBASE:체력 / 15 , MAXBASE:체력)
	BASE:기력 = MIN(BASE:기력 + MAXBASE:기력 / 5 , MAXBASE:기력)
	BASE:MASTER:체력 = MIN(BASE:MASTER:체력 + MAXBASE:MASTER:체력 / 10 , MAXBASE:MASTER:체력)
	BASE:MASTER:기력 = MIN(BASE:MASTER:기력 + MAXBASE:MASTER:기력 / 5 , MAXBASE:MASTER:기력)
	BASE:MASTER:정력 = MIN(BASE:MASTER:정력 + MAXBASE:MASTER:정력 / 8 , MAXBASE:MASTER:정력)
ELSE
	BASE:체력 = MIN(BASE:체력 + MAXBASE:체력 / 8 , MAXBASE:체력)
	BASE:기력 = MIN(BASE:기력 + MAXBASE:기력 / 3 , MAXBASE:기력)
	BASE:MASTER:체력 = MIN(BASE:MASTER:체력 + MAXBASE:MASTER:체력 / 8 , MAXBASE:MASTER:체력)
	BASE:MASTER:기력 = MIN(BASE:MASTER:기력 + MAXBASE:MASTER:기력 / 3 , MAXBASE:MASTER:기력)
	BASE:MASTER:정력 = MIN(BASE:MASTER:정력 + MAXBASE:MASTER:정력 / 4 , MAXBASE:MASTER:정력)
ENDIF
IF TFLAG:193 == -1
	TIMES SOURCE:환락 , 0.50
	TIMES SOURCE:달성 , 0.50
	TIMES SOURCE:사랑 , 0.50
ELSEIF TFLAG:193 == 0
	TIMES SOURCE:환락 , 1.00
	TIMES SOURCE:달성 , 1.00
	TIMES SOURCE:사랑 , 1.00
ELSE
	TIMES SOURCE:환락 , 2.00
	TIMES SOURCE:달성 , 2.00
	TIMES SOURCE:사랑 , 2.00
ENDIF
TIME += 60
CALL 붉은실발동
EXP:MASTER:데이트경험 ++
EXP:TARGET:데이트경험 ++
CALL 만복도상승(MASTER,"주식")
CALL 만복도상승(TARGET,"주식")

SETBIT TFLAG:일일일회, 1
회복후체력1 = BASE:MASTER:체력
회복후기력1 = BASE:MASTER:기력
회복후체력2 = BASE:TARGET:체력
회복후기력2 = BASE:TARGET:기력
DRAWLINE
PRINTFORML 체력+{ABS(회복후체력1 - 회복전체력1)}(%CALLNAME:MASTER%)
PRINTFORML 기력+{ABS(회복후기력1 - 회복전기력1)}(%CALLNAME:MASTER%)
PRINTFORML 체력+{ABS(회복후체력2 - 회복전체력2)}(%CALLNAME:TARGET%)
PRINTFORML 기력+{ABS(회복후기력2 - 회복전기력2)}(%CALLNAME:TARGET%)
RETURN 1

;-------------------------------------------------
;실행 가능 판정
;-------------------------------------------------
;도시락
@COM_ABLE615
;일괄관리
SIF GLOBAL_COMABLE(615)
	RETURN RESULT
;상대가 없다
SIF !TARGET
	RETURN 0
;데이트중 이외
SIF !CHK_DATENOW(CFLAG:12)
	RETURN 0
;데이트 도중
SIF (CFLAG:현재위치 % 100) % 10
	RETURN 0
SIF !OBENTO(CFLAG:MASTER:현재위치)
	RETURN 0
;먹었던 바로 직후
SIF !TIME_PROGRESS(TCVAR:312)
	RETURN 0
SIF GETBIT(TFLAG:일일일회, 1)
	RETURN 0
;수면중
SIF CFLAG:수면
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 1

;FileName_DATE_CMN.ERB ------------------------------- Rev1. 00
;도시락을 먹을 수 있는 장소
;CALL		USER
;ARG		현재위치
;RETURN		먹을 수 있다면 1
;COMMENT	장소를 추가했을 경우, CASE를 변경의 일
;-----------------------------------------------------------
@OBENTO(ARG)
#FUNCTION
IF !AT_HOME(MASTER)
	SELECTCASE ARG
		CASE 지장길, 수양버들아래, 안개의호수, 현무의늪, 큰두꺼비의연못, 절경의언덕, 산의호수, 천계
			RETURNF 1
		;10=桜並木の鳥居
		CASE 무명의언덕, 고목의큰나무, 복숭아숲～모래사장, 10
			RETURNF 1
	ENDSELECT
ENDIF
RETURNF 0

path: ERB/커맨드관련/COMF/외출계/COMF615 도시락을먹는다_t.ERB


code:
;-------------------------------------------------
;진찰을 받는다
;외출 커맨드
;-------------------------------------------------
@COM616
PRINTL 「오늘은 무슨 일일까?」
$LOOP
DRAWLINE
PRINTL [0] 피로가 풀리지 않아서…
PRINTL [1] 왠지 모르게 피곤해지고…
;SIF MAXBASE:MASTER:정력 - BASE:MASTER:정력 > 500
;	PRINTL [2] 그녀가 격렬해서…
PRINTL [3] 요즘 그녀와 매너리즘에 빠져서…
SIF TALENT:MASTER:청소중독
	PRINTL [4] 청소가 너무 즐거워서 그만둘 수가 없습니다
;PRINTL [5] 선생님의 일을 생각하면(자) 자지가 쑥갓응입니다
;PRINTL [5] 선생님의 일을 생각하면(자) 가랑이가 근질근질 합니다
;SIF GETBIT(TALENT:MASTER:성별 & 1)
;PRINTL [10] 자지의 고민
DRAWLINE
PRINTL [100] 그만둔다
INPUT
IF RESULT == 0
	CALL REMEDY(RAND:3)
	PRINTL 치료비로 \3000 청구되었습니다
	IF MONEY < 3000
		DOWNBASE:MASTER:체력 -= 100
		DOWNBASE:MASTER:기력 -= 200
		TIME += 180
		TFLAG:193 = 1
	ELSE
		MONEY -= 3000
	ENDIF
ELSEIF RESULT == 1
	;에이린의 사모 연모로 양화
	CALL REMEDY(RAND:2 + 3 + TALENT:(NO:72):3 * 2 + TALENT:(NO:72):8)
	PRINTL 치료비로 \3000 청구되었습니다
	IF MONEY < 3000
		DOWNBASE:MASTER:체력 -= 100
		DOWNBASE:MASTER:기력 -= 200
		TIME += 180
		TFLAG:193 = 1
	ELSE
		MONEY -= 3000
	ENDIF
ELSEIF RESULT == 3
	PRINTL 「이런 건 어떨까」
	DRAWLINE
	$SELL
	FOR LOCAL, 12,17
		SIF ITEMNAME:LOCAL == ""
			CONTINUE
		PRINTFORM [{LOCAL}] %ITEMNAME:LOCAL% ￥{(ITEMPRICE:LOCAL) / 5}
		IF ITEM:LOCAL
			PRINTL 　(매진)
		ELSE
			PRINTL
		ENDIF
	NEXT
	PRINTL
	PRINTL [100] 돌아온다
	$LOOP2
	INPUT
	IF RESULT >= 12 && RESULT <= 17
		CALL BUY_ABLE(RESULT, 0,1,80)
		GOTO SELL
	ELSEIF RESULT == 100
		RETURN 0
	ELSE
		GOTO LOOP2
	ENDIF
ELSEIF RESULT == 4 && TALENT:MASTER:청소중독
	IF FLAG:850 == 0
		PRINTL 「집안일을 좋아해? 좋은 일 아냐」
		PRINTL 「그런 게 아니라고?」
		PRINTW 「다른 일이 손에 잡히지 않아? 그건 곤란하군」
		PRINTW …
		PRINTW 「으―음, 역시 좀 이상하네」
		PRINTW ……
		PRINTW 「성적으로 흥분!?」
		PRINTW …………
		PRINTL 「도대체 뭘 하면 이렇게 되는 거야」
		PRINTL 「희귀한 증례를 보여 주었으니까 이번은 공짜로 해도 좋아」
		PRINTL 「앞으로는 이상한 청소는 하지 말도록」
		PRINTFORMW %CALLNAME:MASTER%의 [청소중독]이 나았다
		TALENT:MASTER:청소중독 = 0
		FLAG:850 ++
	ELSE
		PRINTL 「또 그걸 재발시켰다고?」
		PRINTFORML 「정말이지… 이번에는 ￥{FLAG:850 * 10000} 받을게」
		IF MONEY < FLAG:850 * 10000
			PRINTL 「치료비가 부족한가 보네」
			PRINTL 「돈이 모이면 다시 와」
		ELSE
			PRINTW …
			PRINTW ……
			PRINTW …………
			PRINTL 「네, 끝」
			PRINTL 「또 재발시키는 일이 있으면 더 많이 지불하도록 할거니깐」
			PRINTFORMW %CALLNAME:MASTER%의 [청소중독]이 나았다
			TALENT:MASTER:청소중독 = 0
			MONEY -= FLAG:850 * 10000
			FLAG:850 ++
		ENDIF
	ENDIF
ELSEIF RESULT == 6
	IF GETBIT(CFLAG:72:기정사실, 1) && EXP:72:펠라티오경험 > 10 && !TALENT:72:키스미경험
		IF TALENT:72:기분 == -1
			PRINTFORMDW %조사처리(CALLNAME:72,"는")% 한숨을 쉬자 %마스터를% 안쪽으로 안내했다
			PRINTFORMDW %조사처리(CALLNAME:72,"는")% 기분나쁜 듯 옆머리를 쓸어올리면서 %CALLNAME:MASTER%의 음경을 물어 넣었다
			PRINTFORMDW %조사처리(CALLNAME:72,"는")% 정액을 입안에 받아낸 후 양치질을 하고 손을 씻어 진찰하러 돌아갔다
			MAXBASE:MASTER:정력 +=1*3;;정력증가
			EXP:MASTER:사정경험 ++
			BASE:MASTER:정력 = MAX(BASE:MASTER:정력 - 200, 0)
			EXP:72:펠라티오경험 ++
			EXP:72:정음경험 ++
			IF TALENT:MASTER:농후정액
				PRINTFORMW %CALLNAME:MASTER%의 [농후정액]이 나았다
				TALENT:MASTER:농후정액 = 0
			ENDIF
			PRINTL 치료비로 \10000 청구되었습니다
			MONEY -= 10000
		ELSE
			PRINTFORMDW 에이린에게 바쁘니까 다음에라며 가볍게 다루어져 버렸다…
			TALENT:72:기분 --
		ENDIF
	ELSE
		PRINTFORMDW %조사처리(CALLNAME:72,"는")% 감사의 말을 하며 %마스터를% 안쪽으로 안내했다
		PRINTW …
		PRINTW ……
		PRINTW ………
		PRINTFORMDW %마스터는% 헌혈에 협력하게 되었다…
		BASE:MASTER:체력 = MIN(MAXBASE:MASTER:체력, BASE:MASTER:체력 - 400)
		TALENT:72:기분 = 1
	ENDIF
;ELSEIF RESULT == 10
;	CALL PENIS_CHANGE
ELSEIF RESULT == 100
	RETURN -1
ELSE
	GOTO LOOP
ENDIF
SETBIT TFLAG:일일일회, 3
TIME += 30
RETURN 1

@REMEDY(ARG)
SELECTCASE ARG
CASE 0
	LOCALS = 맛있는 밥
	BASE:MASTER:체력 = MIN(MAXBASE:MASTER:체력, BASE:MASTER:체력 + 500)
CASE 1
	LOCALS = 영양의 주사
	BASE:MASTER:체력 = MIN(MAXBASE:MASTER:체력, BASE:MASTER:체력 + 750)
CASE 2
	LOCALS = 불끈불끈Z
	BASE:MASTER:체력 = MAXBASE:MASTER:체력
CASE 3
	LOCALS = 계절원―
	BASE:MASTER:기력 = MIN(MAXBASE:MASTER:기력, BASE:MASTER:기력 + 500)
CASE 4
	LOCALS = 진정의 주사
	BASE:MASTER:기력 = MIN(MAXBASE:MASTER:기력, BASE:MASTER:기력 + 750)
CASE 5
	LOCALS = 이마로 체온
	BASE:MASTER:기력 = MIN(MAXBASE:MASTER:기력, BASE:MASTER:기력 + 900)
CASE 6
	LOCALS = 껴안는다
	BASE:MASTER:기력 = MIN(MAXBASE:MASTER:기력, BASE:MASTER:기력 + 1000)
ENDSELECT

PRINTFORMW 에이린은 %LOCALS%에 의한 치료를 실시했다
SIF ARG == 2
	PRINTL 후아 싫어해 개, 만 타 응!
SIF ARG >= 0 && ARG <= 2
	PRINTW 체력이 회복했다
SIF ARG >= 3 && ARG <= 6
	PRINTW 기력이 회복했다
[SKIPSTART]
@PENIS_CHANGE
$TINKO_LOOP
PRINTL [1] 동심에 돌아가고 싶기 때문에 아이 같은 자지로 해 주세요
PRINTL [2] 방탕에 빠지지 않게 데츄 해 주세요
PRINTL [3] 보통 자지로 해 주세요
PRINTL [4] 밤길을 걸을 때 무서워서 빛나는 듯하게 해 주세요
PRINTL [5] 여자아이를 히-히-말하게 하고 싶어서 진주 넣어 주세요
PRINTL [6] 자신을 가질 수 있도록(듯이) 크게 해 주세요
;PRINTL [7] 오크를 동경하고 있기 때문에 드릴같이 해 주세요 그리고 엘프를 찾으러 갑니다
;PRINTL [8] 피임이라든지 귀찮아 응으로 정관 수술 부탁합니다
INPUT
SIF TALENT:MASTER:형상 == RESULT
	GOTO TINKO_LOOP
	LOCAL = RESULT
SELECTCASE RESULT
	CASE 1
		IF GETBIT(CFLAG:72:기정사실, 1) && CFLAG:72:2 > 5000
			PRINTFORMW 에이린은 그러한 것도 개미일까와 생각한 것 같은…
		ELSE
			PRINTFORMW 에이린은 경직된 얼굴로 진심인지 (들)물어 왔다…
	CASE 2
		IF GETBIT(CFLAG:72:기정사실, 1) && CFLAG:72:2 > 5000
			PRINTFORMW 에이린은 다시 생각하도록(듯이) 갔다왔다…
		ELSE
	CASE 3
		PRINTFORMW 에이린은 보통이 제일이라면 기분 좋게 맡아 주었다
	CASE 4
		PRINTFORMW 에이린은 이성인과 컨택 하는 듯한 상태로, 아무것도 그런 곳을 번뜩거릴 수 없어도 괜찮지 않을까라고 (들)물어 왔다…
	CASE 5
		PRINTFORMW 에이린은 그런 것은
[SKIPEND]

;-------------------------------------------------
;실행 가능 판정
;-------------------------------------------------
;진찰
@COM_ABLE616
;일괄관리
SIF GLOBAL_COMABLE(616)
	RETURN RESULT
;에이린이 없으면 안 돼
SIF CFLAG:MASTER:현재위치 != CFLAG:72:현재위치
	RETURN 0
;데이트중
SIF CHK_DATENOW(CFLAG:TARGET:데이트중) && TARGET > 0
	RETURN 0
;데이트 도중
SIF (CFLAG:현재위치 % 100) % 10
	RETURN 0
SIF CFLAG:MASTER:현재위치 != 영원정
	RETURN 0
SIF GETBIT(TFLAG:일일일회, 3)
	RETURN 0
;수면중
SIF CFLAG:MASTER:수면
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 1

path: ERB/커맨드관련/COMF/외출계/COMF616 진찰받는다_t.ERB


code:
;-------------------------------------------------
;과자가게에 권한다
;외출 커맨드
;-------------------------------------------------
@COM617
#DIM 회복전체력1
#DIM 회복전기력1
#DIM 회복후체력1
#DIM 회복후기력1
#DIM 회복전체력2
#DIM 회복전기력2
#DIM 회복후체력2
#DIM 회복후기력2

회복전체력1 = BASE:MASTER:체력
회복전기력1 = BASE:MASTER:기력
회복전체력2 = BASE:TARGET:체력
회복전기력2 = BASE:TARGET:기력

CALL SUC_FAIL()
SELECTCASE TFLAG:193
	;실패
	CASE -1
		SIF TFLAG:192 == -2
			RETURN 0
	;대성공
	CASE 1
		CALL 붉은실발동
		EXP:MASTER:회화경험 ++
		EXP:MASTER:데이트경험 ++
		EXP:TARGET:데이트경험 ++
	;성공
	CASE 0
ENDSELECT
TFLAG:98 += 1

DOWNBASE:기력 += 10

;고정으로 획득하는 소스
SOURCE:환락 = 800
SOURCE:사랑 = 300

;ABL:욕망을 본다
IF ABL:욕망 <= 1
	SOURCE:환락 += (ABL:욕망 * 60)
ELSEIF ABL:욕망 <= 3
	SOURCE:환락 += 300 + (ABL:욕망 * 60)
ELSEIF ABL:욕망 <= 5
	SOURCE:환락 += 600 + (ABL:욕망 * 60)
ELSEIF ABL:욕망 <= 8
	SOURCE:환락 += 900 + (ABL:욕망 * 75)
ELSEIF ABL:욕망 <= 11
	SOURCE:환락 += 1500 + (ABL:욕망 * 75)
ELSE
	SOURCE:환락 += 2500 + (ABL:욕망 * 30)
ENDIF

;ABL:봉사정신을 본다
IF ABL:봉사정신 <= 1
	SOURCE:사랑 += (ABL:봉사정신 * 30)
ELSEIF ABL:봉사정신 <= 3
	SOURCE:사랑 += 200 + (ABL:봉사정신 * 30)
ELSEIF ABL:봉사정신 <= 5
	SOURCE:사랑 += 500 + (ABL:봉사정신 * 30)
ELSEIF ABL:봉사정신 <= 8
	SOURCE:사랑 += 700 + (ABL:봉사정신 * 45)
ELSEIF ABL:봉사정신 <= 11
	SOURCE:사랑 += 1200 + (ABL:봉사정신 * 45)
ELSE
	SOURCE:사랑 += 1800 + (ABL:봉사정신 * 20)
ENDIF

;호감 도를 본다
IF CFLAG:2 <= 500
	SOURCE:환락 += CFLAG:2
ELSEIF CFLAG:2 <= 5000
	SOURCE:환락 += 750 + (CFLAG:2 - 500) / 3
ELSE
	SOURCE:환락 += 2000 + (CFLAG:2 - 5000) / 4
ENDIF
SIF SOURCE:환락 < 0
	SOURCE:환락 = 0

SOURCE:수동 = 200 + 100 * ABL:순종
SOURCE:정복 = 200 + 100 * ABL:새드끼

IF TFLAG:193 == -1
	TIMES SOURCE:환락 , 0.10
	TIMES SOURCE:사랑 , 0.10
	TIMES SOURCE:정복 , 0.50
	TIMES SOURCE:수동 , 0.50
ELSEIF TFLAG:193 == 0
	TIMES SOURCE:환락 , 1.00
	TIMES SOURCE:사랑 , 1.00
	TIMES SOURCE:정복 , 1.00
	TIMES SOURCE:수동 , 1.00
ELSE
	TIMES SOURCE:환락 , 2.00
	TIMES SOURCE:사랑 , 1.50
	TIMES SOURCE:정복 , 2.00
	TIMES SOURCE:수동 , 2.00
	SIF TALENT:15 < 1
		TALENT:15 = 1
ENDIF

BASE:체력 = MIN(BASE:체력 + MAXBASE:체력 / 25 , MAXBASE:체력)
BASE:기력 = MIN(BASE:기력 + MAXBASE:기력 / 4 , MAXBASE:기력)
BASE:MASTER:체력 = MIN(BASE:MASTER:체력 + MAXBASE:MASTER:체력 / 25 , MAXBASE:MASTER:체력)
BASE:MASTER:기력 = MIN(BASE:MASTER:기력 + MAXBASE:MASTER:기력 / 4 , MAXBASE:MASTER:기력)

TIME += 60
CALL 붉은실발동
EXP:MASTER:데이트경험 ++
EXP:TARGET:데이트경험 ++
CALL 만복도상승(TARGET,"디저트")
CALL 만복도상승(MASTER,"디저트")
MONEY -= 3000
회복후체력1 = BASE:MASTER:체력
회복후기력1 = BASE:MASTER:기력
회복후체력2 = BASE:TARGET:체력
회복후기력2 = BASE:TARGET:기력
DRAWLINE
PRINTFORML 체력+{ABS(회복후체력1 - 회복전체력1)}(%CALLNAME:MASTER%)
PRINTFORML 기력+{ABS(회복후기력1 - 회복전기력1)}(%CALLNAME:MASTER%)
PRINTFORML 체력+{ABS(회복후체력2 - 회복전체력2)}(%CALLNAME:TARGET%)
PRINTFORML 기력+{ABS(회복후기력2 - 회복전기력2)}(%CALLNAME:TARGET%)
RETURN 1

;-------------------------------------------------
;실행 가능 판정
;-------------------------------------------------
;과자가게로 이끈다
@COM_ABLE617
;일괄관리
SIF GLOBAL_COMABLE(617)
	RETURN RESULT
;상대가 없다
SIF !TARGET
	RETURN 0
;데이트중 이외
SIF !CHK_DATENOW(CFLAG:12)
	RETURN 0

IF !AT_HOME(MASTER)
	SIF CFLAG:MASTER:현재위치 != 광장
		RETURN 0
ELSE
	SIF CFLAG:MASTER:현재위치 != 217
		RETURN 0
ENDIF

;먹었던 바로 직후
SIF !TIME_PROGRESS(TCVAR:디저트공복시각)
	RETURN 0
SIF MONEY < 3000
	RETURN 0
;수면중
SIF CFLAG:수면
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 1

path: ERB/커맨드관련/COMF/외출계/COMF617 과자가게로부른다_t.ERB


code:
;-------------------------------------------------
;꽃집
;외출 커맨드
;-------------------------------------------------
@COM618
PRINTL 소지 아이템
CALL LIST_ITEM, "원예용품"
PRINTL
PRINTL
PRINTFORML 구입할 아이템을 선택해주세요 소지금:\\{MONEY}

CALL SHOW_ITEM, "원예용품"
PRINT [100] - 돌아간다
$INPUT_LOOP
INPUT
IF RESULT == 100
	TIME += 10
	RETURN -1
ELSEIF ITEM_TYPES( RESULT, "원예용품" )
	IF RESULT == 150 && ITEM:원예용품일식
		PRINTW 이미 가지고 있습니다
	ELSE
		CALL BUY_ABLE, RESULT
	ENDIF
	PRINTL
	RESTART
ELSE
	GOTO INPUT_LOOP
ENDIF
TIME += 10

;-------------------------------------------------
;실행 가능 판정
;-------------------------------------------------
;꽃집
@COM_ABLE618
;일괄관리
SIF GLOBAL_COMABLE(618)
	RETURN RESULT

IF !AT_HOME(MASTER)
	SIF CFLAG:MASTER:현재위치 != 상가거리
		RETURN 0
ELSE
	SIF CFLAG:MASTER:현재위치 != 215
		RETURN 0
ENDIF
;수면중
SIF CFLAG:수면
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 1

path: ERB/커맨드관련/COMF/외출계/COMF618 꽃집_t.ERB


code:
;-------------------------------------------------
;술집
;외출 커맨드
;-------------------------------------------------
@COM619
#DIM 탁주판매개수
#DIM 청주판매개수
#DIM 오니고로시판매개수
#DIM 재고증가수
;RAND를 걸치는 것은 일일일회
IF FLAG:마지막에술집에방문한날 != DAY
	;연회 당일은 술이 품귀상태로
	IF FLAG:개시일 == DAY
		탁주판매개수 = RAND:(4 + FLAG:술집매각율플래그 * 4)
		청주판매개수     = RAND:(3 + FLAG:술집매각율플래그 * 3)
		오니고로시판매개수   = RAND:(2 + FLAG:술집매각율플래그 * 2)
	ELSE
		탁주판매개수 = RAND:(6 + FLAG:술집매각율플래그 * 8)
		청주판매개수     = RAND:(5 + FLAG:술집매각율플래그 * 6)
		오니고로시판매개수   = RAND:(4 + FLAG:술집매각율플래그 * 4)
	ENDIF
	;누적 구입액에 응해도 재고가 증가
	IF FLAG:술집에서누적구입액
		재고증가수 = MAX(FLAG:술집에서누적구입액 / 100000 - 2, 0) + 1
		탁주판매개수 += RAND:재고증가수
		청주판매개수     += RAND:재고증가수
		오니고로시판매개수   += RAND:재고증가수
	ENDIF
ENDIF

IF FLAG:마지막에술집에방문한날 == 0
	PRINTL 「첫 손님이구나? 서비스할게」
ELSEIF DAY - FLAG:마지막에술집에방문한날 > 7
	PRINTL 「오랜만이네요―, 천천히 둘러봐요」
ELSEIF FLAG:마지막에술집에방문한날 == DAY
	PRINTL 「어서 오세요, 응, 어머머? 웬 일이야?」
ELSEIF FLAG:술집에서누적구입액 > 1000000 && FLAG:개시일 != DAY
	PRINTFORML 「어서 오세요, %마스터를% 위해서 널럴하게 들여놨지」
ELSEIF FLAG:개시일 == DAY
	PRINTL 「때가 안 좋았네. 아까 온 손님이 거의 다 사갔어」
ELSEIF 탁주판매개수 + 청주판매개수 + 오니고로시판매개수 > 30
	PRINTL 「어서 오세요, 좋은 술이 많이 있어―」
ELSE
	PRINTL 「어서 오십시오―」
ENDIF
SIF FLAG:술집매입강화
	PRINTFORML 「현재 매입 강화 캠페인 중이야. 물건 조달해주면 톡톡히 쳐줄게」
$INPUT_LOOP2
PRINTL [0] - 술을 산다
PRINTL [1] - 술을 판다
PRINTL [100] - 돌아간다
$INPUT_LOOP
INPUT

IF RESULT == 0
	CALL 술구입(탁주판매개수, 청주판매개수, 오니고로시판매개수)
	GOTO INPUT_LOOP2
ELSEIF RESULT == 1
	CALL 술도매(탁주판매개수, 청주판매개수, 오니고로시판매개수)
	GOTO INPUT_LOOP2
ELSEIF RESULT != 100
	GOTO INPUT_LOOP
ENDIF

FLAG:마지막에술집에방문한날 = DAY
IF FLAG:술집누적전매이익 > 3000000 && FLAG:술집매각율플래그 == 2
	FLAG:술집매각율플래그 = 3
	PRINTL
	PRINTL 가게에서 나가려 했는데 불러 세웠다
	PRINTL 「두둑하게 술을 조달해줘서 정말 살겠어∼, 앞으로는 매입 금액 더 쳐줄게」
ELSEIF FLAG:술집누적전매이익 > 1000000 && FLAG:술집매각율플래그 == 1
	FLAG:술집매각율플래그 = 2
	PRINTL
	PRINTL 가게에서 나가려 했는데 불러 세웠다
	PRINTL 「두둑하게 술을 조달해줘서 살겠어―, 매입 금액 더 쳐줄 테니 더 조달해주지 않을래?」
ELSEIF FLAG:술집누적전매이익 > 100000 && FLAG:술집매각율플래그 == 0
	FLAG:술집매각율플래그 = 1
	PRINTL
	PRINTL 가게에서 나가려 했는데 불러 세웠다
	PRINTL 「늘 술을 조달해줘서 고마워. 사례로 다음 번부터 매입 금액 더 쳐줄게―」
ELSE
	PRINTL 「또 오세요―」
ENDIF
TIME += 10
RETURN -1

@술구입(탁주판매개수, 청주판매개수, 오니고로시판매개수)
#DIM REF 탁주판매개수
#DIM REF 청주판매개수
#DIM REF 오니고로시판매개수
#DIM 판매개수
#DIM 소지가능개수
#DIM 가격인하율
#DIM 선택한상품
#DIM 구입개수
#DIM 분명하게샀다

;가격인하율을 20%로 설정(정가의 80%로 판매)
가격인하율 = 20

;각각의 술의 최대 소지 가능수를 80으로 설정(80개 이상 가지고 있다고(면) 그 술은 살 수 없습니다)
소지가능개수 = 80

PRINTL
IF FLAG:마지막에술집에방문한날 == 0
	PRINTL 「처음 왔으니 서비스야―」
	;가격인하율을 40%로 설정(정가의 60%로 판매)
	가격인하율 = 40
ELSE
	PRINTL 「어느 거나 사서 손해 없어―」
ENDIF
$INPUT_LOOP2
PRINTFORML 소지금 :\\%TOSTR(MONEY,"#,###")%
FOR LOCAL, 51,54
	SELECTCASE LOCAL
		CASE 51
			판매개수 = 탁주판매개수
		CASE 52
			판매개수 = 청주판매개수
		CASE 53
			판매개수 = 오니고로시판매개수
	ENDSELECT
	LOCALS = [{LOCAL}]%ITEMNAME:(LOCAL), 8, LEFT%	(\\%TOSTR(ITEMPRICE:(LOCAL) * (100 - 가격인하율) / 100), 5, RIGHT%)
	IF ITEMPRICE:LOCAL
		PRINTFORM %LOCALS%
		IF 판매개수 == 0
			SETCOLOR C_RED
			PRINT 	(다 팔림)
			RESETCOLOR
		ELSE
			PRINTFORM 	(재고:{판매개수})
		ENDIF
		IF ITEM:LOCAL >= 소지가능개수
			SETCOLOR C_YELLOW
			PRINTFORM \t소지:{ITEM:LOCAL, 2, RIGHT}
			FONTSTYLE 1
			PRINTL 구입 불가
			RESETCOLOR
			FONTREGULAR
		ELSE
			PRINTFORML \t소지:{ITEM:LOCAL, 2, RIGHT}
		ENDIF
	ENDIF
NEXT
PRINT [100] - 돌아간다
$INPUT_LOOP
INPUT
IF RESULT == 100
	RETURN -1
ELSEIF RESULT >= 51 && RESULT <= 53

	;나중에 다시 한번 INPUT를 사용하기 위해 지금의 값을 퇴피
	선택한상품 = RESULT

	;선택한 번호에 따라판매개수를 격납
	SELECTCASE 선택한상품
		CASE 51
			판매개수 = 탁주판매개수
		CASE 52
			판매개수 = 청주판매개수
		CASE 53
			판매개수 = 오니고로시판매개수
	ENDSELECT

	;품절의 경우
	IF 판매개수 == 0
		PRINTL
		PRINTL 「미안해요. 다 팔렸어요」

	;최대 소지 가능수이상의 술을 소지하고 있는 경우
	ELSEIF ITEM:선택한상품 >= 소지가능개수
		PRINTL 「손님, 그만큼 있으면 충분하잖아요? 오히려 우리한테 물건을 대주라―」

	;그 이외의경우 개수를 묻는다
	ELSE
		구입개수 = 0
		$BUY_LOOP
		PRINTFORML %ITEMNAME:선택한상품%를 몇 개 살까…
		PRINTFORML 사는 수：{구입개수}
		PRINTFORML [ -10]　[  -1]　[  +1]　[ +10]
		PRINTBUTTON "[전부] ", 999
		PRINTBUTTON " [리셋]", -999
		PRINTL
		PRINTBUTTON "[산다]", 0
		INPUT
		IF RESULT
			IF RESULT == 999
				구입개수 =판매개수
			ELSEIF RESULT == -999
				구입개수 = 0
			ELSE
				구입개수 = LIMIT(구입개수 + RESULT, 0,판매개수)
			ENDIF
			CLEARLINE 6
			GOTO BUY_LOOP
		ENDIF
		;0이하의 값을 입력했을 경우
		IF 구입개수 <= 0
			PRINTL
			PRINTW 역시 사는 건 그만두자
			GOTO INPUT_LOOP2
		;재고의 수보다 많이 구입하려고 했을 경우
		ELSEIF 판매개수 < 구입개수
			PRINTW 재고가 그렇게 많진 않은 것 같다
			GOTO INPUT_LOOP2
		;최대 소지 가능수보다 많이 구입하려고 했을 경우
		ELSEIF 소지가능개수 < ITEM:선택한상품 + 구입개수
			PRINTW 그렇게나 가져봤자 처치곤란하다
			GOTO INPUT_LOOP2
		ENDIF
		CALL BUY_ABLE(선택한상품, 0, 구입개수, 가격인하율)
		분명하게샀다 = RESULT
		IF FLAG:술집에서누적구입액 > 100000 && !RAND:9 && 분명하게샀다
			PRINTL
			PRINTL 「항상 사줘서 고마워, 이건 덤이야」
			PRINTFORMW ―덤으로 %ITEMNAME:510%를 받았다-
			ITEM:510 ++
		ELSE
			PRINTL
			PRINTW 「매번 감사―」
		ENDIF
		IF 분명하게샀다
			FLAG:술집에서누적구입액 += ITEMPRICE:선택한상품 * (100 - 가격인하율) / 100 * 구입개수
			;판매개수를 감소시키는 처리
			SELECTCASE 선택한상품
				CASE 51
					탁주판매개수 -= 구입개수
				CASE 52
					청주판매개수     -= 구입개수
				CASE 53
					오니고로시판매개수   -= 구입개수
			ENDSELECT
		ENDIF
	ENDIF
	GOTO INPUT_LOOP2
ELSE
	GOTO INPUT_LOOP2
ENDIF
RETURN

@술도매(탁주판매개수, 청주판매개수, 오니고로시판매개수)
#DIM REF 탁주판매개수
#DIM REF 청주판매개수
#DIM REF 오니고로시판매개수
#DIM 가격인하율
#DIM 소지수
#DIM 선택한상품
#DIM 매각개수
#DIM 매각액

SELECTCASE FLAG:술집매각율플래그
	CASE 0
		;가격인하율을 90%로 설정(정가의 10%로 구입해 준다)
		가격인하율 = 90
	CASE 1
		;가격인하율을 85%로 설정(정가의 15%로 구입해 준다)
		가격인하율 = 85
	CASE 2
		;가격인하율을 80%로 설정(정가의 20%로 구입해 준다)
		가격인하율 = 80
	CASE 3
		;가격인하율을 70%로 설정(정가의 30%로 구입해 준다)
		가격인하율 = 70
ENDSELECT
SIF FLAG:술집매입강화
	가격인하율 -= 20
PRINTL
PRINT 「
IF FLAG:술집매입강화
	PRINT 매입 강화 캠페인중이니까 특별히
ELSEIF FLAG:술집매각율플래그
	PRINT 고객이니까 특별히
ENDIF
PRINTFORML 정가의 {100 - 가격인하율}\%로 살게」
PRINTL 무엇을 팔까…
$INPUT_LOOP2
PRINTFORML 소지금 :\\%TOSTR(MONEY,"#,###")%
FOR LOCAL, 51,54
	LOCALS = [{LOCAL}]%ITEMNAME:(LOCAL)%
	IF ITEMPRICE:LOCAL
		PRINTFORML %LOCALS%	(\\{ITEMPRICE:(LOCAL) * (100 - 가격인하율) / 100})  	(소지:{ITEM:LOCAL})
	ENDIF
NEXT
PRINT [100] - 돌아간다
$INPUT_LOOP
INPUT
IF RESULT == 100
	RETURN -1
ELSEIF RESULT >= 51 && RESULT <= 53
	;선택한 번호에 따라 소지수를 격납
	소지수 = ITEM:RESULT

	;소지 0의 경우
	IF 소지수 == 0
		PRINTL
		PRINTL 물건이 없기에 팔고 싶어도 팔 수 없다
	;소지 0으로 없는 경우 개수를 묻는다
	ELSE
		;다시 한번 INPUT를 사용하기 때문에 지금의 값을 퇴피
		선택한상품 = RESULT
		매각개수 = 0
		$SELL_LOOP
		PRINTFORML %ITEMNAME:선택한상품%를 몇 개 팔까…
		PRINTFORML 파는 수：{매각개수}
		PRINTFORML [-100]　[ -10]　[  -1]　[  +1]　[ +10]　[+100]
		PRINTBUTTON "[전부] ", 999
		PRINTBUTTON " [리셋]", -999
		PRINTL
		PRINTBUTTON "[판다]", 0
		INPUT
		IF RESULT
			IF RESULT == 999
				매각개수 = 소지수
			ELSEIF RESULT == -999
				매각개수 = 0
			ELSE
				매각개수 = LIMIT(매각개수 + RESULT, 0, 소지수)
			ENDIF
			CLEARLINE 6
			GOTO SELL_LOOP
		ENDIF
		IF 매각개수 <= 0
			PRINTL
			PRINTL 역시 파는 건 그만두자
			GOTO INPUT_LOOP2
		ENDIF
		IF 매각개수 > 소지수
			PRINTL
			PRINTL 그렇게나 많이 가지고 있지 않다
			GOTO INPUT_LOOP2
		ENDIF

		;1개당의 매각값을 산출
		매각액 = ITEMPRICE:선택한상품 * (100 - 가격인하율) / 100
		ITEM:선택한상품 -= 매각개수
		MONEY += 매각액 * 매각개수
		;가게는 정가의 80%로 판매
		FLAG:술집누적전매이익 += ITEMPRICE:선택한상품 * (가격인하율 - 20) / 100 * 매각개수
		PRINTFORMW %ITEMNAME:선택한상품%를 ￥{매각액*매각개수}로 {매각개수}개 매각했습니다
		;판매개수를 증가시키는 처리
		SELECTCASE 선택한상품
		CASE 51
			탁주판매개수 += 매각개수
		CASE 52
			청주판매개수     += 매각개수
		CASE 53
			오니고로시판매개수   += 매각개수
		ENDSELECT
	ENDIF
	GOTO INPUT_LOOP2
ELSE
	GOTO INPUT_LOOP
ENDIF
;-------------------------------------------------
;실행 가능 판정
;-------------------------------------------------
;술집
@COM_ABLE619
;일괄관리
SIF GLOBAL_COMABLE(619)
	RETURN RESULT
IF !AT_HOME(MASTER)
	SIF CFLAG:MASTER:현재위치 != 술집소로
		RETURN 0
ELSE
	SIF CFLAG:MASTER:현재위치 != 211
		RETURN 0
ENDIF
;수면중
SIF CFLAG:수면
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 1

path: ERB/커맨드관련/COMF/외출계/COMF619 술집_t.ERB


code:
;대본가게
@COM620
#DIM CHARISMA
PRINTFORML 소지금:\\{MONEY}
PRINTL 「임차료는 모두\3000입니다」
$C_LOOP
PRINT [1] 전술지도서

IF ITEM:101
	PRINTFORML (%마스터가% 빌리고 있다)
ELSEIF GETBIT (FLAG:1011, 1)
	PRINTL (대출중)
ELSE
	PRINTL
ENDIF
PRINT [2] 조리지도서
IF ITEM:102
	PRINTFORML (%마스터가% 빌리고 있다)
ELSEIF GETBIT (FLAG:1011, 2)
	PRINTL (대출중)
ELSE
	PRINTL
ENDIF
PRINT [3] 사전
IF ITEM:103
	PRINTFORML (%마스터가% 빌리고 있다)
ELSEIF GETBIT (FLAG:1011, 3)
	PRINTL (대출중)
ELSE
	PRINTL
ENDIF
PRINT [4] 단편소설
IF ITEM:104
	PRINTFORML (%마스터가% 빌리고 있다)
ELSEIF GETBIT (FLAG:1011, 4)
	PRINTL (대출중)
ELSE
	PRINTL
ENDIF
PRINT [5] 연주지도서
IF ITEM:105
	PRINTFORML (%마스터가% 빌리고 있다)
ELSEIF GETBIT (FLAG:1011, 5)
	PRINTL (대출중)
ELSE
	PRINTL
ENDIF
PRINT [7] 생선도감
IF ITEM:107
	PRINTFORML （%마스터가% 빌리고 있다）
ELSEIF GETBIT (FLAG:1011,7)
	PRINTL （대출중）
ELSE
	PRINTL
ENDIF
PRINT [8] 채집가이드
IF ITEM:108
	PRINTFORML （%마스터가% 빌리고 있다）
ELSEIF GETBIT (FLAG:1011,8)
	PRINTL （대출중）
ELSE
	PRINTL
ENDIF
PRINT [9] 약학교본
IF ITEM:109
	PRINTFORML （%마스터가% 빌리고 있다）
ELSEIF GETBIT (FLAG:1011,9)
	PRINTL （대출중）
ELSE
	PRINTL
ENDIF

IF !CHARISMA
	PRINTFORML [100] 카리스마 지불
ELSE
	PRINTFORML [100] 현금지불
ENDIF
DRAWLINE
PRINTL [0] 그만둔다

$LOOP
INPUT
IF ITEM:101 || ITEM:102 || ITEM:103 || ITEM:104 || ITEM:105 || ITEM:107 || ITEM:108 || ITEM:109
	PRINTL 「2권 이상의 대출은 할 수 없습니다」
	RETURN -1
ELSEIF RESULT == 100
	IF !CHARISMA
		PRINTFORMW 「카리스마로 지불해?」
		PRINTFORMW 「그러니까……자 10 카리스마? 그리고 좋습니다……」
		IF MONEY:2 < 10
			PRINTFORMW 카리스마가 부족하다
			GOTO C_LOOP
		ENDIF
	ELSE
		PRINTFORMW 「아, 돈으로 지불해 줍니다」
	ENDIF
	PRINTFORML
	CHARISMA = !CHARISMA
	GOTO C_LOOP
ELSE
	IF INRANGE(RESULT, 1, 9)
		IF GETBIT (FLAG:1011, RESULT)
			PRINTL 「대출중입니다」
		ELSE
			SETBIT FLAG:1011, RESULT
			IF !CHARISMA
				MONEY -= 3000
			ELSE
				MONEY:2 -= 10
			ENDIF
			ITEM:(100 + RESULT) = 1
			FLAG:대본반환기한 = 8
			TFLAG:193 = 1
			PRINTFORML %ITEMNAME:(100 + RESULT)%를 빌렸다
			RETURN 1
		ENDIF
	ELSEIF RESULT == 0
		RETURN -1
	ELSE
		GOTO LOOP
	ENDIF
ENDIF
TIME += 30

;대본소
@COM_ABLE620
;일괄관리
SIF GLOBAL_COMABLE(620)
	RETURN RESULT
IF !AT_HOME(MASTER)
	SIF CFLAG:MASTER:현재위치 != 상가거리
		RETURN 0
ELSE
	SIF CFLAG:MASTER:현재위치 != 213
		RETURN 0
ENDIF

SIF MONEY < 3000
	RETURN 0
;수면중
SIF CFLAG:수면
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 1

@BOOKLEND

FOR LOCAL,1,10
	SIF LOCAL == 6	;[그림연극]을 패스
		CONTINUE
	SIF !ITEM:(100 + LOCAL) && !RAND:3
		INVERTBIT FLAG:1011, LOCAL
NEXT
SIF FLAG:대본반환기한
	FLAG:대본반환기한 -= 1
IF FLAG:대본반환기한 == 1
	ITEM:101 = 0
	ITEM:102 = 0
	ITEM:103 = 0
	ITEM:104 = 0
	ITEM:105 = 0
	ITEM:107 = 0
	ITEM:108 = 0
	ITEM:109 = 0
	PRINTL 반환 기한이 왔으므로, 빌리고 있던 책을 돌려주었습니다.
ENDIF

path: ERB/커맨드관련/COMF/외출계/COMF620 스즈나안_t.ERB


code:
;-------------------------------------------------
;낚시를 한다
;외출 커맨드
;-------------------------------------------------
@COM621
#DIM 잡힌것
#DIM 낚시파워
#DIM 낚시파워M
#DIM 낚시파워T
#DIM 잡히지않는보정
#DIM 낚시운
#DIM CHARA
#DIM 상대자가낚시했다
#DIM 活性
#DIM FISHED, 2

상대자가낚시했다 = 0
;참가자 판정
IF FLAG:데이트상대
	CHARA = FLAG:데이트상대
ELSEIF TARGET && ABL:TARGET:친밀 >= 5 - TALENT:TARGET:낚시Lv
	CHARA = TARGET
ELSE
	CHARA = 0
ENDIF
;불쌍해；；
IF CHARA == [[와카사기히메]] || TARGET == [[와카사기히메]]
	PRINTFORMW 그녀 앞에서 낚시를 하는 것은 가혹한 것이다…
	RETURN -1
ENDIF

;季節ごとの活性
SELECTCASE DAY:2
	CASE 1  ;春
		活性 = 10
	CASE 2  ;夏
		活性 = 7
	CASE 3  ;秋
		活性 = 10
	CASE 4  ;冬
		活性 = 3
ENDSELECT
;時刻ごとの活性補正
SELECTCASE TIME:2
	CASE 1, 4  ;早朝と昼下がり
		活性 += 10
	CASE 5, 6  ;夕と夜
		活性 += 5
ENDSELECT
;天気による活性
SELECTCASE TIME:5
	CASE 3, 6 ;曇、霧雨
		活性 += 5
	CASE 4, 5 ;雨
		活性 += 7
	CASE 8 TO 13 ;雪系統はマイナス
		活性 -= 3
ENDSELECT
;虹が出てるときは、雨の後で水が濁っているだろう
SELECTCASE TIME:7
	CASE 1
		活性 += 7
ENDSELECT

;일일로―50~+50, 아이템이나 기원등으로 보정을 걸치는 일도 가능
낚시운 = FLAG:일일이벤트 / 10 - 50

;釣り技能で活性を察知
IF TALENT:MASTER:낚시Lv >= 3
	SETCOLOR C_L_GRAY
	SELECTCASE (活性 + 낚시운)
		CASE IS <= 10
			PRINTFORML （魚影がほとんど見えない…）
		CASE 11 TO 30
			PRINTFORML （魚影はまばらだ…）
		CASE 31 TO 50
			PRINTFORML （魚影はそれなりに見える…）
		CASE 51 TO 65
			PRINTFORML （魚影はかなり多く見える…）
		CASEELSE
			PRINTFORML （魚影がひしめき、水面がざわついている…）
	ENDSELECT
	RESETCOLOR
ENDIF

;낚시의 묘미 (참다운 즐거움)은 대기 시간이라고 들었다
FOR LOCAL, 0, 5
	REUSELASTLINE  \@ CHARA ?  %조사처리(CALLNAME:CHARA,"과")% 함께 #  \@낚시중. %". " * LOCAL%
	TWAIT 150, 1
NEXT
;낚시파워설정
낚시파워M = 10 + TALENT:MASTER:낚시Lv * 15 + 낚시운 + 잡히지않는보정
SIF CHARA
	낚시파워T = 10 + TALENT:CHARA:낚시Lv * 15 + 잡히지않는보정

;(낚시파워)%로 잡힌다
FISHED:0 = RAND:100
IF FISHED < 낚시파워M
	낚시파워= 낚시파워M
	$FISH
	;잡히면(자) 보정 리셋
	잡히지않는보정 = 0
	;낚시파워로 잡히는 것이 바뀐다
	SELECTCASE MIN(RAND:낚시파워, 100)
	CASE IS < 5
		;목재
		잡힌것 = 610
	CASE IS < 10
		;모리야부적
		잡힌것 = 131
	CASE 51 TO 60
		;개구리
		잡힌것 = 706
	CASE 10 TO 50, 61 TO 89
		FISHED:1 = (FISHED:0 + 10) / 10 * 10
		잡힌것 = FISHING_CATCH(FISHED:1)
	CASE IS < 95
		잡힌것 = FISHING_CATCH(95)
	CASEELSE
		잡힌것 = FISHING_CATCH(100)
		;보너스
		IF 상대자가낚시했다
			EXP:CHARA:낚시경험 ++
		ELSE
			EXP:MASTER:낚시경험 ++
		ENDIF
	ENDSELECT
	;아이템에 추가
	SELECTCASE 잡힌것
		CASE 1 TO 10, 22 TO 23
			SIF ITEM:물고기 < 99
				ITEM:물고기 ++
		CASE 11 TO 20, 24 TO 60
			SIF ITEM:맛있는물고기 < 99
				ITEM:맛있는물고기 ++
		CASE 21
			;SIF ITEM:로터 < 99
				ITEM:로터 ++
		CASEELSE
			SIF ITEM:잡힌것 < 99
			ITEM:잡힌것 ++
	ENDSELECT
;CHARA가 낚시했을 경우
ELSEIF CHARA && RAND:100 < 낚시파워T
	낚시파워= 낚시파워T
	상대자가낚시했다 = 1
	GOTO FISH
;잡히지 않았으면 보정 가산, 잡힐 때까지 축적해 나간다
ELSE
	잡히지않는보정 += 10
	잡힌것 = 0
ENDIF

;기본의 경험
EXP:MASTER:낚시경험 ++

SIF ITEM:107 && !RAND:2
	EXP:MASTER:낚시경험 ++

SIF CHARA
	EXP:CHARA:낚시경험 ++
;낚시한 사람의 추가 경험
IF 잡힌것
	IF 상대자가낚시했다
		EXP:CHARA:낚시경험 ++
	ELSE
		EXP:MASTER:낚시경험 ++
	ENDIF
ENDIF
;데이트경험
IF CHK_DATENOW(CFLAG:CHARA:데이트중) && CHARA > 0
	CALL 붉은실발동
	EXP:MASTER:데이트경험 ++
	EXP:CHARA:데이트경험 ++
ENDIF

;당신이 낚시하면(자) 체력 감소
SIF 잡힌것 && !상대자가낚시했다
	DOWNBASE:MASTER:체력 = 50
DOWNBASE:MASTER:기력 = 100

;참가자의 소스, 적당하게 설정하고 있습니다, 동일 커맨드 제한을 받지 않기 때문에 조심스럽게, 다만 잡히지 않으면 매회 반감
IF CHARA > 0
	;고정으로 획득하는 소스
	SOURCE:CHARA:환락 = 400
	SOURCE:CHARA:사랑 = 50

	;ABL:순종을 본다
	IF ABL:순종 <= 1
		SOURCE:CHARA:환락 += (ABL:순종 * 40)
		SOURCE:CHARA:사랑 += (ABL:친밀 * 20)
	ELSEIF ABL:순종 <= 3
		SOURCE:CHARA:환락 += 200 + (ABL:순종 * 40)
		SOURCE:CHARA:사랑 += 50 + (ABL:친밀 * 30)
	ELSEIF ABL:순종 <= 5
		SOURCE:CHARA:환락 += 500 + (ABL:순종 * 40)
		SOURCE:CHARA:사랑 += 200 + (ABL:친밀 * 30)
	ELSEIF ABL:순종 <= 8
		SOURCE:CHARA:환락 += 750 + (ABL:순종 * 60)
		SOURCE:CHARA:사랑 += 300 + (ABL:친밀 * 40)
	ELSEIF ABL:순종 <= 11
		SOURCE:CHARA:환락 += 1000 + (ABL:순종 * 60)
		SOURCE:CHARA:사랑 += 600 + (ABL:친밀 * 40)
	ELSE
		SOURCE:CHARA:환락 += 1600 + (ABL:순종 * 30)
		SOURCE:CHARA:사랑 += 1000 + (ABL:친밀 * 20)
	ENDIF
	SIF CHARA != FLAG:데이트상대
		SOURCE:CHARA:사랑 = 0

	;호감 도는 보지 않는다

	SOURCE:CHARA:수동 = 200 + 100 * ABL:순종
	SOURCE:CHARA:정복 = 200 + 100 * ABL:새드끼

	;낚시질의 성과를 본다
	SELECTCASE 잡힌것
	;잡히지 않는다
	CASE 0
		TIMES SOURCE:CHARA:환락 , 0.50
		TIMES SOURCE:CHARA:수동 , 0.50
		TIMES SOURCE:CHARA:정복 , 0.50
	;목재, 모리야부적
	CASE 610, 131
		TIMES SOURCE:CHARA:환락 , 0.10
		TIMES SOURCE:CHARA:수동 , 0.50
		TIMES SOURCE:CHARA:정복 , 0.50
		SOURCE:CHARA:달성 = 20
	;황어, 아마고, 송어 , 이와나
	CASE 1 TO 10, 22
		TIMES SOURCE:CHARA:환락 , 1.00
		TIMES SOURCE:CHARA:수동 , 1.00
		TIMES SOURCE:CHARA:정복 , 1.00
		SOURCE:CHARA:달성 = 100
	;개구리
	CASE 706
		IF CHARA == [[치르노]]
			TIMES SOURCE:CHARA:환락 , 2.00
			TIMES SOURCE:CHARA:수동 , 2.00
			TIMES SOURCE:CHARA:정복 , 2.00
		ELSEIF CHARA == [[스와코]]
			TIMES SOURCE:CHARA:환락 , 0.50
			TIMES SOURCE:CHARA:수동 , 0.50
			TIMES SOURCE:CHARA:정복 , 0.50
		ELSE
			TIMES SOURCE:CHARA:환락 , 1.25
			TIMES SOURCE:CHARA:수동 , 1.00
			TIMES SOURCE:CHARA:정복 , 1.00
		ENDIF
		SOURCE:CHARA:달성 = 150
	;술, 빙어, 니지마스, 캄르치
	CASE 11 TO 13, 15 TO 20, 24 TO 30
		TIMES SOURCE:CHARA:환락 , 1.50
		TIMES SOURCE:CHARA:수동 , 1.50
		TIMES SOURCE:CHARA:정복 , 1.50
		SOURCE:CHARA:달성 = 200
	;칠성장어
	CASE 14
		;みすちー
		IF CHARA == 28
			TIMES SOURCE:CHARA:환락 , 2.00
			TIMES SOURCE:CHARA:수동 , 2.00
			TIMES SOURCE:CHARA:정복 , 2.00
		ELSE
			TIMES SOURCE:CHARA:환락 , 1.50
			TIMES SOURCE:CHARA:수동 , 1.50
			TIMES SOURCE:CHARA:정복 , 1.50
		ENDIF
		SOURCE:CHARA:달성 = 250
	;로터
	CASE 21
		SOURCE:CHARA:욕정 = 200
	;シーラカンス，タキタロウ，얼음의비늘, 용의손톱，龍涎香
	CASE 23, 31, 609, 606, 643
		TIMES SOURCE:CHARA:환락 , 2.00
		TIMES SOURCE:CHARA:수동 , 2.00
		TIMES SOURCE:CHARA:정복 , 2.00
		SOURCE:CHARA:달성 = 300
	ENDSELECT
ENDIF

;시간경과
TIME += 20

;플래그 인도
TFLAG:SELECTCOM분기 = 잡힌것
TFLAG:SOURCE등에영향을주지않는SELECTCOM분기 = CHARA + (상대자가낚시했다 * 1000)

RETURN 1

;-------------------------------------------------
;실행 가능 판정
;-------------------------------------------------
;낚시를 한다
@COM_ABLE621
;일괄관리
SIF GLOBAL_COMABLE(621)
	RETURN RESULT
;낚싯대가 없다
SIF !ITEM:낚싯대
	RETURN 0
;거점에서는 할 수 없다(가짜)
SIF !FISHING_SPOT(CFLAG:MASTER:현재위치)
	RETURN 0
;데이트중은 데이트상대와 만
SIF FLAG:데이트상대 && TARGET != FLAG:데이트상대
	RETURN 0
;수면중
SIF CFLAG:MASTER:수면
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 1

;FileName_DATE_CMN.ERB ------------------------------- Rev1. 00
;낚시가 생기는 장소
;CALL		USER
;ARG		현재위치
;RETURN		할 수 있다면 1
;COMMENT	장소를 추가했을 경우, CASE를 변경의 일 현재 KR판에만 있음
;-----------------------------------------------------------
@FISHING_SPOT(ARG)
#FUNCTION
IF !IN_HOME(ARG)
	SELECTCASE ARG
		CASE 안개의호수
			RETURNF 1
		CASE 삼도천
			RETURNF 2
		CASE 현무의늪
			RETURNF 3
		CASE 큰두꺼비의연못
			RETURNF 4
		CASE 구천의폭포
			RETURNF 5
		CASE 산의호수
			RETURNF 6
		CASE 지옥의깊은길
			RETURNF 7
	ENDSELECT
ELSE
	SELECTCASE ARG
		CASE 347;霧の湖
			RETURNF 1
		CASE 602;賽の河原
			RETURNF 2
		CASE 703;玄武の沢
			RETURNF 3
		CASE 708;大蝦蟇の池
			RETURNF 4
		CASE 806;滝の裏
			RETURNF 5
		CASE 814;湖畔（山の湖）
			RETURNF 6
		CASE 945;川岸
			RETURNF 7
	ENDSELECT
ENDIF
RETURNF 0

;-------------------------------------------------
;釣れたもの判定の拡張　場所、季節、時刻に応じて可変
;ARG　…ランクに対応
;-------------------------------------------------
@FISHING_CATCH(ARG)
#FUNCTION

SELECTCASE ARG
	CASE 20
		SELECTCASE FISHING_SPOT(CFLAG:MASTER:현재위치)
			CASE 1, 4 ;버들치
				RETURNF 5
			CASE 2, 7 ;피라미
				RETURNF 6
			CASE 3, 5, 6 ;황어
				RETURNF 1
		ENDSELECT
	CASE 30
		SELECTCASE FISHING_SPOT(CFLAG:MASTER:현재위치)
			CASE 1, 2, 6 ;까나리
				RETURNF 7
			CASE 3, 5 ;송어
				RETURNF 2
			CASE 4 ;개구리
				RETURNF 706
			CASE 7 ;피라니아
				RETURNF 8
		ENDSELECT
	CASE 40
		SELECTCASE FISHING_SPOT(CFLAG:MASTER:현재위치)
			CASE 1, 2, 6 ;둑중개
				RETURNF 16
			CASE 3, 5 ;산천어ちゃん
				RETURNF 3
			CASE 4 ;개구리
				RETURNF 706
			CASE 7 ;가아
				RETURNF 9
		ENDSELECT
	CASE 50
		SELECTCASE FISHING_SPOT(CFLAG:MASTER:현재위치)
			CASE 1, 6 ;ブルーギル
				RETURNF 10
			CASE 2, 7 ;アユ（春夏）or피라미（秋冬）
				IF DAY:2 < 3
					RETURNF 17
				ELSE
					RETURNF 6
				ENDIF
			CASE 3, 5 ;곤들매기
				RETURNF 4
			CASE 4 ;개구리
				RETURNF 706
		ENDSELECT
	CASE 70
		SELECTCASE FISHING_SPOT(CFLAG:MASTER:현재위치)
			CASE 1, 4, 6 ;ナマズorワカサギ（冬）
				IF DAY:2 == 4
					RETURNF 12
				ELSE
					RETURNF 18
				ENDIF
			CASE 2, 7 ;ウナギ
				RETURNF 19
			CASE 3, 5 ;ニジマスor연어（秋）orワカサギ（冬）
				IF DAY:2 == 3
					RETURNF 11
				ELSEIF DAY:2 == 4
					RETURNF 12
				ELSE
					RETURNF 13
				ENDIF
		ENDSELECT
	CASE 80
		SELECTCASE FISHING_SPOT(CFLAG:MASTER:현재위치)
			CASE 1 ;ヒメマス
				RETURNF 20
			CASE 2, 4, 6 ;コイ
				RETURNF 22
			CASE 3, 5 ;ヤツメウナギ
				RETURNF 14
			CASE 7 ;チョウザメ
				RETURNF 24
		ENDSELECT
	CASE 90
		SELECTCASE FISHING_SPOT(CFLAG:MASTER:현재위치)
			CASE 1 ;クニマス
				RETURNF 25
			CASE 2 ;ソウギョ
				RETURNF 26
			CASE 3 ;カムルチー
				RETURNF 15
			CASE 4 ;ヘラブナ
				RETURNF 27
			CASE 5 ;イトウ
				RETURNF 28
			CASE 6 ;オオクチバス
				RETURNF 29
			CASE 7 ;ピラルクー
				RETURNF 30
		ENDSELECT
	CASE 95
		SELECTCASE  FISHING_SPOT(CFLAG:MASTER:현재위치)
			CASE 1 ;タキタロウ（新月の日）or로터
				IF MOON_WATCH(DAY:3) == 1
					RETURNF 31
				ELSE
					RETURNF 21
				ENDIF
			CASE 2 ;シーラカンス
				RETURNF 23
			CASEELSE
				;로터
				RETURNF 21
		ENDSELECT
	CASE 100
		SELECTCASE FISHING_SPOT(CFLAG:MASTER:현재위치)
			CASE 1, 6 ;龍涎香or얼음의비늘（冬）
				IF DAY:2 == 4
					RETURNF 609
				ELSE
					RETURNF 643
				ENDIF
			CASE 2, 3, 5, 7 ;용의손톱or얼음의비늘（冬）
				IF DAY:2 == 4
					RETURNF 609
				ELSE
					RETURNF 606
				ENDIF
			CASE 4 ;개구리
				RETURNF 706
		ENDSELECT
ENDSELECT
RETURNF 0

;-----------------------------------------------------------
;釣った魚のナンバーを名称に変換
;コレクションリストにも利用するため、로터は除外しておく
;ARG　…釣れたものナンバーに対応
;-----------------------------------------------------------
@FISH_NAME(ARG)
#FUNCTIONS

SELECTCASE ARG
	CASE 1
		RETURNF "황어"
	CASE 2
		RETURNF "송어"
	CASE 3
		RETURNF "산천어"
	CASE 4
		RETURNF "곤들매기"
	CASE 5
		RETURNF "버들치"
	CASE 6
		RETURNF "피라미"
	CASE 7
		RETURNF "까나리"
	CASE 8
		RETURNF "피라니아"
	CASE 9
		RETURNF "가아"
	CASE 10
		RETURNF "ブルーギル"
	CASE 11
		RETURNF "연어"
	CASE 12
		RETURNF "ワカサギ"
	CASE 13
		RETURNF "ニジマス"
	CASE 14
		RETURNF "ヤツメウナギ"
	CASE 15
		RETURNF "カムルチー"
	CASE 16
		RETURNF "둑중개"
	CASE 17
		RETURNF "アユ"
	CASE 18
		RETURNF "ナマズ"
	CASE 19
		RETURNF "ウナギ"
	CASE 20
		RETURNF "ヒメマス"
	CASE 22
		RETURNF "コイ"
	CASE 23
		RETURNF "シーラカンス"
	CASE 24
		RETURNF "チョウザメ"
	CASE 25
		RETURNF "クニマス"
	CASE 26
		RETURNF "ソウギョ"
	CASE 27
		RETURNF "ヘラブナ"
	CASE 28
		RETURNF "イトウ"
	CASE 29
		RETURNF "オオクチバス"
	CASE 30
		RETURNF "ピラルクー"
	CASE 31
		RETURNF "タキタロウ"
	CASEELSE
		RETURNF ""
ENDSELECT

;-------------------------------------------------
;釣れた魚のサイズを一般的サイズ内でランダムに出力する変数
;ARG　…釣れたものナンバーに対応
;-------------------------------------------------
@FISHING_SIZE(ARG)
#FUNCTION

LOCAL = 0
SELECTCASE ARG
	CASE 1 ;황어
		LOCAL = MAX(RAND:(56), 25)
	CASE 2 ;송어
		LOCAL = MAX(RAND:(40), 18)
	CASE 3 ;산천어ちゃん
		LOCAL = MAX(RAND:(50), 25)
	CASE 4 ;곤들매기
		LOCAL = MAX(RAND:(70), 20)
	CASE 5 ;버들치
		LOCAL = MAX(RAND:(18), 10)
	CASE 6 ;피라미
		LOCAL = MAX(RAND:(18), 10)
	CASE 7 ;까나리
		LOCAL = MAX(RAND:(36), 15)
	CASE 8 ;피라니아
		LOCAL = MAX(RAND:(61), 15)
	CASE 9 ;가아
		LOCAL = MAX(RAND:(300), 120)
	CASE 10 ;ブルーギル
		LOCAL = MAX(RAND:(28), 12)
	CASE 11 ;연어
		LOCAL = MAX(RAND:(100), 65)
	CASE 12 ;ワカサギ
		LOCAL = MAX(RAND:(18), 8)
	CASE 13 ;ニジマス
		LOCAL = MAX(RAND:(105), 35)
	CASE 14 ;ヤツメウナギ
		LOCAL = MAX(RAND:(65), 45)
	CASE 15 ;カムルチー
		LOCAL = MAX(RAND:(115), 80)
	CASE 16 ;둑중개
		LOCAL = MAX(RAND:(18), 10)
	CASE 17 ;アユ
		LOCAL = MAX(RAND:(35), 25)
	CASE 18 ;ナマズ
		LOCAL = MAX(RAND:(80), 50)
	CASE 19 ;ウナギ
		LOCAL = MAX(RAND:(120), 85)
	CASE 20 ;ヒメマス
		LOCAL = MAX(RAND:(55), 20)
	CASE 21 ;로터
		LOCAL = 5
	CASE 22 ;コイ
		LOCAL = MAX(RAND:(115), 55)
	CASE 23 ;シーラカンス
		LOCAL = MAX(RAND:(200), 100)
	CASE 24 ;チョウザメ
		LOCAL = MAX(RAND:(301), 100)
	CASE 25 ;クニマス
		LOCAL = MAX(RAND:(45), 30)
	CASE 26 ;ソウギョ
		LOCAL = MAX(RAND:(140), 70)
	CASE 27 ;ヘラブナ
		LOCAL = MAX(RAND:(62), 25)
	CASE 28 ;イトウ
		LOCAL = MAX(RAND:(220), 90)
	CASE 29 ;オオクチバス
		LOCAL = MAX(RAND:(55), 25)
	CASE 30 ;ピラルクー
		LOCAL = MAX(RAND:(400), 250)
	CASE 31; タキタロウ
		LOCAL = MAX(RAND:(550), 300)
ENDSELECT
RETURNF LOCAL

;-------------------------------------------------
;釣果の説明、消去行数の関係で５行に統一
;ARG　…釣れたものナンバーに対応
;-------------------------------------------------
@FISHING_EXPLANATION(ARG)
IF INRANGE(ARG, 11, 20) || INRANGE(ARG, 24, 60)
	CALL COLOR_PRINT("　　☆맛있는물고기☆", "C_YELLOW")
ELSE
	PRINTFORM 　　◇普通の魚◇
ENDIF
PRINTL
SELECTCASE ARG
	CASE 1 ;황어
		PRINTFORML 　　体長30cm～50cm前後、雑食性。背側は焦茶色、春には黒と赤の婚姻色を表す
		PRINTFORML 　　２～４年で成魚となり、群れて生活する。小骨が多く、臭いあり
		PRINTFORML 　　河川の上流から下流、はたまた山上湖にも生息。水質汚染や酸性水にも強い
		PRINTFORML
		PRINTFORML
	CASE 2 ;송어
		PRINTFORML 　　体長20cm～30cm前後、小魚や昆虫食。산천어とよく似た姿だが、側面に散在する朱点で区別できる
		PRINTFORML 　　サツキマスの陸封型（海や湖に降りず、河川で一生を終える）で、年間を通して水温の低い場所を好む
		PRINTFORML 　　漢字は「雨子」「雨魚」などが当てられ、梅雨時～初夏など雨がちな時期によく釣れる
		PRINTFORML 　　また、「甘い（美味しいの意）魚」という意味の異称を持つほど、美味である
		PRINTFORML
	CASE 3 ;산천어ちゃん
		PRINTFORML 　　体長は～40cm前後、ミミズや昆虫食。背側は緑がかった黄褐色、若いうちは小判状の斑紋模様（パーマーク）を示す
		PRINTFORML 　　サクラマスの陸封型（海や湖に降りず、河川で一生を終える）で、河川上流の冷水域に生息
		PRINTFORML 　　寿命は３～４年。生活スタイルの似る곤들매기よりも下流に生息し「棲み分け」を行う
		PRINTFORML 　　非常に警戒心が強く釣るのは難しいとされ、「渓流の女王」の異名をとる
		PRINTFORML
	CASE 4 ;곤들매기
		PRINTFORML 　　体長20～60cm前後、貪欲な肉食性で、蛇すら食した事例も。背側は緑褐色や灰色で、地域によって様々
		PRINTFORML 　　河川最上流の冷水域に生息し、生活スタイルの似る산천어よりも上流に暮らし「棲み分け」を行う
		PRINTFORML 　　かつては他の연어の仲間と同様に海に降りる生活をしていたようだが、本種はとりわけ冷水を好む性質ゆえ、
		PRINTFORML 　　氷河期の終焉とともに陸封型の生活に固定され、地方地方の河川ごとに分断された結果、多様な亜種が生まれたようだ
		PRINTFORML
	CASE 5 ;버들치
		PRINTFORML 　　体長～15cm程度、雑食性。体色は黄褐色で鱗が小さく、ぬめりが強いため「アブラ」の名をとる
		PRINTFORML 　　河川の中流から上流域、山地の湖沼など、低水温を好む。群れて産卵をする習性をもつ
		PRINTFORML 　　多少の苦みがあるが、天ぷらやマリネ、南蛮漬けなどに
		PRINTFORML
		PRINTFORML
	CASE 6 ;피라미
		PRINTFORML 　　体長～15cm程度、背側は灰白色、繁殖期のオスには青緑とピンクの鮮やかな婚姻色が表れる
		PRINTFORML 　　河川下流から中流域、湖沼などに生息。水質汚染や河川改修による環境変化に強い
		PRINTFORML 　　草食寄りの雑食性。甘露煮や天ぷら、なれずしなどに
		PRINTFORML
		PRINTFORML
	CASE 7 ;까나리
		PRINTFORML 　　キンブナは体長15cm程度、ギンブナは30cm程度まで成長。キンブナは黄褐色の体色をしている
		PRINTFORML 　　雑食性で、河川下流域や湖沼などに生息するコイの仲間
		PRINTFORML 　　マブナとも言われるが、これはゲンゴロウブナ（ヘラブナ）と他の까나리を区別するための総称
		PRINTFORML 　　까나리の仲間の分類は非常に難しく、いまだに定説はないとされている
		PRINTFORML
	CASE 8 ;피라니아
		PRINTFORML 　　アマゾン川などの熱帯地方に生息する、鋭い歯を持つ肉食魚の総称
		PRINTFORML 　　現地の言葉で「歯のある魚」の意味で、小型種では15cmほど、大型種では60cmほどまで成長する
		PRINTFORML 　　「人食い魚」は誇張されたイメージで、実態は臆病。群れて暮らすが、群れ全体が興奮状態に入ると集団で攻撃を開始する
		PRINTFORML 　　肉はタンパク質豊富で美味。地底の地熱で温かい川に適応しているのだろうか？
		PRINTFORML
	CASE 9 ;가아
		PRINTFORML 　　堅牢な硬鱗（ガノイン鱗）を持つ、魚類の中でも原始的なスタイルの魚。北アメリカ大陸の原産
		PRINTFORML 　　細長い胴体と細長く伸びるアゴを持ち、種によっては３ｍに迫る巨体にまで成長することがある
		PRINTFORML 　　近年日本の某河川でも排水（温水）によって定着しているのではないかと言われている
		PRINTFORML 　　地底の地熱で温かい川に適応しているようだが……誰が持ち込んだのだろう？
		PRINTFORML
	CASE 10 ;ブルーギル
		PRINTFORML 　　北米大陸が原産の魚。体長～25cm程度。湖沼や流れの緩やかな河川に生息する
		PRINTFORML 　　パンフィッシュ（フライパンに収まりやすい魚）と俗称されるグループの魚だけあり、丸っこく扁平な形をしている
		PRINTFORML 　　非常に悪食で、何でも食べる貪欲さが移植先の水棲生物に大きな影響を与えてしまい、駆除の対象となっている
		PRINTFORML 　　幻想郷にも分布を広げているのだとすれば、大変なことになりそうだ
		PRINTFORML
	CASE 11 ;연어
		PRINTFORML 　　体長70cm程度まで成長。川でふ化した稚魚は3～4月に海に降り、非常に長い距離を回遊しながら成長する
		PRINTFORML 　　４～６年ほど海で過ごした後、秋ごろに生まれた川を遡上し、湧水のある河床に産卵、一生を終える
		PRINTFORML 　　遡上してくる産卵期の個体の食味は卵や白子の形成によって脂・旨味が落ちており、沖合で漁獲されたものに及ばない
		PRINTFORML 　　……海のないはずの幻想郷で、一体何処でこれほどの成熟を遂げてくるのだろうか？
		PRINTFORML
	CASE 12 ;ワカサギ
		PRINTFORML 　　体長は成魚で15cm程度。湖沼やそれに注ぐ河川に生息。一部の個体は海に降りることもある
		PRINTFORML 　　冬に産卵を行う一年魚で、冬・氷上の釣りがイメージされるが、実際は一年を通して釣れる
		PRINTFORML 　　冷水を好むとともに、存外に塩分や水質汚染にも強く、大量産卵をすることで多くの仲間を増やす繁殖力を併せ持つ
		PRINTFORML 　　食用にも向いており、諏訪湖をはじめ多くの湖に移植された
		PRINTFORML
	CASE 13 ;ニジマス
		PRINTFORML 　　別名レインボートラウト。体長は一般には40cm程度だが、湖などに住むものは80cm～1ｍ近くまで成長する
		PRINTFORML 　　産卵しても１回では死なず、数年にわたって産卵を行うことができる
		PRINTFORML 　　寄生虫がいないと言われ、刺身でも食べられる珍しい淡水魚で、繁殖期のオスには虹色の婚姻色が発現する
		PRINTFORML 　　日本に移入されたのは明治時代。幻想郷が大結界で隔てられた後の事だが……どこから入ってきたのだろう？
		PRINTFORML
	CASE 14 ;ヤツメウナギ
		PRINTFORML 　　体長は50～60cm。無顎類（アゴを持たない魚類）と呼ばれる極めて原始的な魚類で「生きた化石」
		PRINTFORML 　　体側に7対のエラの穴があり、それが目のように見えることから「八つ目」の名を持つ
		PRINTFORML 　　清浄な冷水を好み、降海型と陸封型が存在する。なお、ウナギとは完全に別種である
		PRINTFORML 　　食味は硬くクセはあるが栄養は豊富で、ビタミンＡの一種を多量に含み、夜盲症に絶大な効果があるという
		PRINTFORML
	CASE 15 ;カムルチー
		PRINTFORML 　　別名ライギョ。体長は１ｍ程度まで成長する。水流が緩やかで、水草などが多い環境を好む
		PRINTFORML 　　魚食性だが、時にはカメや水鳥のヒナをも捕食する悪食。水底に潜み、待ち伏せを行う
		PRINTFORML 　　白身で淡白な味わいで、揚げ物や鍋で食される。タイワンドジョウの別称の通り、ユーラシア大陸が原産の外来種
		PRINTFORML 　　「そちら」の方面からやってきた誰かが持ち込んだものだろうか？
		PRINTFORML
	CASE 16 ;둑중개
		PRINTFORML 　　カサゴの仲間の淡水魚で体長15cm程度まで成長。体色は褐色～淡褐色まで変化に富む
		PRINTFORML 　　綺麗な水の河川の瀬や砂利底に潜んでいることが多く、水生昆虫や小魚をエサとする
		PRINTFORML 　　見た目は悪いが、とても美味でよいダシが出る。唐揚げや鍋、甘露煮など、広く料理に用いられる
		PRINTFORML
		PRINTFORML
	CASE 17 ;アユ
		PRINTFORML 　　体長～30cm程度、成魚になると石についた藻類を食べて成長する。１年で寿命を終える魚
		PRINTFORML 　　成魚は縄張りを持ち、侵入したアユを追い払う習性をもつ。これを利用したのが囮のアユを泳がせて釣る「友釣り」
		PRINTFORML 　　本来は川と海を回遊する魚であるが、琵琶湖流域では海の代わりに琵琶湖に降り、遡上して大きく成長するものもいるという
		PRINTFORML 　　海のない幻想郷では、湖での成熟を行っているのだろうか？
		PRINTFORML
	CASE 18 ;ナマズ
		PRINTFORML 　　体長は60cm程度。河川の中～下流域や湖沼、水田や用水路にも生息。夜行性で、発達した口ヒゲで周囲を探る肉食魚
		PRINTFORML 　　古くから、カエルをつけた仕掛けで水面を叩いて誘う「ポカン釣り」という手法で釣り上げられていた
		PRINTFORML 　　なお、地震の予兆としてナマズが暴れる、という俗信があるが、科学的な裏付けはまだ取れていないとされる
		PRINTFORML 　　地下に大ナマズがおり、これが暴れると大地震が起こる、というのも説話として有名……え？　夢に出た？
		PRINTFORML
	CASE 19 ;ウナギ
		PRINTFORML 　　体長は１ｍ程度まで成長する。河川の中～下流域や湖沼に生息。体表はヌルヌルした粘液に覆われる
		PRINTFORML 　　夜行性の肉食魚で、淡水魚に分類されるが、海に降って産卵・ふ化を行う降河回遊を行う
		PRINTFORML 　　深海でふ化し、海で成長してシラスウナギに、その後沿岸から川を遡上して成長を続け、５～１０年で成熟する
		PRINTFORML 　　その姿形から龍の化身、菩薩の遣いになぞらえられることもあり、食すのを禁忌とする地域もあったようだ
		PRINTFORML 　　めっきり外の世界で見なくなったのは、ウナギが幻想入りしているから、なのだろうか？
	CASE 20 ;ヒメマス
		PRINTFORML 　　体長は20～50cm。北米大陸に生息するベニザケの陸封型（海や湖に降りず、河川で一生を終える魚）
		PRINTFORML 　　冷水を好むが、水深が深く冷たい湧水がある湖では暖地でも定着するため、いくつかの湖に移植された
		PRINTFORML 　　体色は灰色で、甲殻類などの動物プランクトンを主食とし、ワカサギとはエサを巡り競合を起こす
		PRINTFORML 　　연어科の魚の中でも最も美味とされるが、すぐに食味が落ちてしまう
		PRINTFORML
	;CASE 21 ;로터

	CASE 22 ;コイ
		PRINTFORML 　　体長は60cm～１ｍ程度まで成長。規模の大きい河川の下流～中流域、湖沼、用水路などに広く生息
		PRINTFORML 　　低水温や汚染に耐える生命力を持つうえ寿命が長く、70～80年も生き延びる個体も存在する
		PRINTFORML 　　水底を吸い混ぜ、およそ口に入るものなら何でも食べる強い雑食性を示す
		PRINTFORML 　　よく見られる体高の高いコイは、中国から大昔に移入されたものであるといわれている
		PRINTFORML 　　琵琶湖水系などに生息するスリムなコイが日本に古来から生息するものと推測され、野鯉と俗称される
	CASE 23 ;シーラカンス
		PRINTFORML 　　体長は１ｍを超える。かつて広く世界中に分布した古代魚で、肉質のヒレを数多く持つ
		PRINTFORML 　　現生の陸上脊椎動物は全てこの系統から進化したと言われており、深海から生存種が発見された生きた化石
		PRINTFORML 　　味は申し訳ないがまずい。カニの肉をさらに薄味にしたような味、ハブラシのような食味、と評されている
		PRINTFORML 　　この個体は淡水性で、外の世界ではとうに絶滅したはずの種と考えられる……幽霊？
		PRINTFORML
	CASE 24 ;チョウザメ
		PRINTFORML 　　体長１ｍ～４ｍにも成長。形はサメに似るが、サメとは別種の古代魚
		PRINTFORML 　　卵を塩漬けにしたものがキャビアであり、各地で乱獲の憂き目にあった
		PRINTFORML 　　かつて日本にも生息していたが、２０世紀初頭に絶滅したという
		PRINTFORML 　　現代は温泉水の熱などを利用し、外国産の品種の養殖も行われている
		PRINTFORML
	CASE 25 ;クニマス
		PRINTFORML 　　体長30cm～40cm。体色は灰色から黒色。ヒメマスの亜種とも言われたが別種の可能性もあるという
		PRINTFORML 　　水深の深い湖に生息し、産卵時は30～40ｍの比較的浅い所まで上がってくるという
		PRINTFORML 　　もとは秋田県の田沢湖にのみ生息した固有種であったが、湖に酸性の河川水を引き入れた事により絶滅
		PRINTFORML 　　しかし、昭和初期に山梨県の西湖に卵が移入されており、その子孫が絶滅宣言から70年後に再発見された
		PRINTFORML 　　どう食べても非常に美味、と評されており、かつては特産品として献上されていた
	CASE 26 ;ソウギョ
		PRINTFORML 　　体長は1.2ｍ程度、高水温や酸欠にも強い、水草や水際の雑草を食する草食性のコイの仲間
		PRINTFORML 　　もとは中国の大河に生息しており、上流で産卵した浮遊性の卵が水流に流されるうちにふ化する特性をもつ
		PRINTFORML 　　卵が海まで流されると死んでしまうため、流れの緩やかで長大な流域面積を持つ河でないと繁殖できない
		PRINTFORML 　　中国では四大家魚と称されるほどポピュラーな魚で、蒸し魚・煮魚・揚げ物・スープなど、広く食べられている
		PRINTFORML
	CASE 27 ;ヘラブナ
		PRINTFORML 　　琵琶湖水系のゲンゴロウブナの変種を品種改良したもので、横から見るとひし形の形をしている
		PRINTFORML 　　体長は30～60cm程度。寿命は長く、数十年生きるものもいるという
		PRINTFORML 　　「釣りは까나리に始まり까나리に終わる」の金言の、「終わりの까나리」が本種であるとされ、難易度が高い
		PRINTFORML 　　季節により別種であるように行動が変わり、水温や天気によって遊泳層（タナ）も変わるうえ、
		PRINTFORML 　　採餌も溶け出したエサを小さく吸い込むだけで、感度の高いウキや竿を用いなければアタリをとるのも難しい
	CASE 28 ;イトウ
		PRINTFORML 　　体長1～1.5ｍ、時に２ｍを超える日本最大の淡水魚。冷水で流れの緩い河川の中流～下流域に生息
		PRINTFORML 　　背は青みがかった褐色。かつては本州北部にも生息していたが、現在は北海道にのみ分布する
		PRINTFORML 　　日本のイトウは연어のように降海するが、幻想郷の本種は外国種のように陸封型の生活を送っているようだ
		PRINTFORML 　　肉は白身で脂が乗って美味しいが、成長が遅いのが難点
		PRINTFORML
	CASE 29 ;オオクチバス
		PRINTFORML 　　体長30～50cm、北米原産の肉食魚で、湖沼や河川の中～下流域に生息
		PRINTFORML 　　自分の体長の半分ほどの魚や、小鳥やカエル、ネズミまで丸呑みにしてみせる貪欲さを持つ
		PRINTFORML 　　スポーツフィッシングの対象として有名だが、移入先で生態系を圧迫するなど問題も多い
		PRINTFORML 　　肉は淡白な白身でクセが無く、ムニエルやフライなどでおいしく頂ける
		PRINTFORML
	CASE 30 ;ピラルクー
		PRINTFORML 　　南米大陸に生息している体長３ｍを超える大魚。魚食性で、丸太のような胴、平たい頭部と尾ひれが特徴
		PRINTFORML 　　世界最大級の淡水魚の一種で、ガーの仲間と並んで古くから姿を変えていない古代魚
		PRINTFORML 　　体色は黒から銀色まで様々で、成魚では後部が赤色に変色する。現地の言葉で「紅い魚」と呼ばれる由縁
		PRINTFORML 　　肉は食べごたえがあり、無数の突起状の骨の通った舌（骨舌）はおろし器、鱗はヤスリや靴べらに、と広く利用された
		PRINTFORML
	CASE 31; タキタロウ
		PRINTFORML 　　体長３ｍ以上にもなるという大怪魚。大型のマスのような姿で口は兎のような三口、下あごが突き出ている
		PRINTFORML 　　冷たく深い湖に生息しているとされ、その実態はいまだ掴めていない謎の魚
		PRINTFORML 　　肉は赤身で多くの脂を含み、簡単には焼けないが非常に美味であるという
		PRINTFORML 　　正体は、全くの新種の魚か、곤들매기の変種か、湖に残留するイトウか、はたまたキングサーモンか
		PRINTFORML 　　その真相は、今も深く冷たい水の中に隠れている……
ENDSELECT

path: ERB/커맨드관련/COMF/외출계/COMF621 낚시한다_t.ERB


code:
;-------------------------------------------------
;홉고블린에의뢰
;외출 커맨드
;-------------------------------------------------
@COM630
PRINTFORML 홉고블린에게 뭔가 부탁하겠습니까?
CALL ASK_M("부탁하지 않는다", 1,"청소 대행",!TFLAG:출장청소,"물주기대행", ITEM:150)
SIF !RESULT
	RETURN -1
IF RESULT == 1
	PRINTFORML 홉고블린에게 %GET_MAPNAME(MAIN_MAP)%의 청소를 부탁하겠습니까?
	CALL PAY_YEN_OR_CM(3000,10)
	SIF !RESULT
		RETURN -1
	PRINTFORML 출장 청소 서비스를 의뢰했습니다.
	TFLAG:출장청소 = 1
ELSEIF RESULT == 2
	PRINTFORML 홉고블린에게 텃밭의 물주기를 부탁하겠습니까? (이번 달까지)
	CALL PAY_YEN_OR_CM(15000,70)
	SIF !RESULT
		RETURN -1
	PRINTFORML 물주기대행을 의뢰했습니다.
	FLAG:물주기대행 = 1
ENDIF
TIME += 5
RETURN 1

@COM_ABLE630
;일괄관리
SIF GLOBAL_COMABLE(630)
	RETURN RESULT
SIF MAIN_MAP == 3
	RETURN 0
SIF CFLAG:MASTER:현재위치 != 350
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
RETURN 1

path: ERB/커맨드관련/COMF/외출계/COMF630 홉고블린에게의뢰_t.ERB


code:
;-------------------------------------------------
;홍마카지노
;외출 커맨드
;-------------------------------------------------
@COM631
TFLAG:카지노입장시카리스마 = MONEY:2
PRINTFORMW %마스터는% 홍마관내의 카지노 시설로 발길을 옮겼다
CALL CASINO_ENTRANCE

SIF NEMUKE() >= 700
	PRINTFORM 졸음이 한계를 맞이하고 있다
PRINTFORMW %마스터는% 카지노를 뒤로 했다
TIME += 10

@COM_ABLE631
;일괄관리
SIF GLOBAL_COMABLE(631)
	RETURN RESULT
LOCAL = 309
SIF MAIN_MAP != 3
	LOCAL = GET_MAP_REPLACEMENT(LOCAL)
SIF CFLAG:MASTER:현재위치 != LOCAL
	RETURN 0
SIF TFLAG:망상중
	RETURN 0
;시간정지중
SIF FLAG:70
	RETURN 0
;시간에 여유가 없다
;활동 한계를 700으로서 그 20분전
SIF NEMUKE() >= 700
	RETURN 0
RETURN 1

path: ERB/커맨드관련/COMF/외출계/COMF631 홍마카지노_t.ERB


code:
;-------------------------------------------------
;대련
;외출 커맨드
;-------------------------------------------------
@COM632
PRINTFORML 전투의 체험판입니다. 해보겠습니까?
CALL ASK_M("안 한다", 1,"엄청 약한 적", 1,"약한 적", 1)
IF !RESULT
	RETURN -1
ELSE
	CALL BATTLE_SYSTEM(RESULT)
	IF RESULT == -1
		PRINTFORMW 졌다!
	ELSEIF RESULT == 1
		PRINTFORMW 이겼다! 장하다!
	ENDIF
ENDIF

@COM_ABLE632
;일괄관리
SIF GLOBAL_COMABLE(632)
	RETURN RESULT
SIF !TFLAG:100
	RETURN 0
;정지 중에는 불가
SIF FLAG:70 == 1
	RETURN 0
;업무중
SIF CFLAG:행동 == 5
	RETURN 0
SIF !전투훈련가능(CFLAG:MASTER:현재위치)
	RETURN 0
;기력 0
SIF BASE:MASTER:기력 <= 0
	RETURN 0
SIF CFLAG:우후후 == 2
	RETURN 0
RETURN 1

path: ERB/커맨드관련/COMF/외출계/COMF632 대련_t.ERB


code:
@COM650
PRINTL 「git도 사용할 수 없는 사람에게 파는 것은 없어」
PRINTW git는 상식의 범주에 들어가는 것 같다
RETURN -1

path: ERB/커맨드관련/COMF/외출계/COMF650 git_t.ERB